

H a n d b u c h

der Programmiersprache

B A S I C - 8 0

d e s M I C R O C O M P U T E R S M C 8 0

VEB Elektronik Gera

Hersteller :
VEB Elektronik Gera
DDR 6500 Gera
Parkstraße 3
Postschließfach III/334

Das vorliegende Handbuch gibt keinerlei Auskunft über Liefermöglichkeiten.

Die technischen Angaben dieser Beschreibung dienen der Information.

Änderungen im Sinne des wissenschaftlich-technischen Fortschrittes behalten wir uns vor.

Ausgabe 11/86

Inhaltsverzeichnis

1.	Einleitung	I 1...5
2.	BASIC-80-Grundlagen	II 1...2
3.	EDITOR BASIC-80	III 1...11
3.1.	Bedienkommandos	
3.2.	Fehlerausschriften	
3.3.	Programmtest	
4.	Anweisungen	IV 1...18
4.1.	Wertzuweisung 'LET'	
4.2.	Die Funktion 'INP'	
4.3.	Ausgabeeweisungen	
4.3.1.	Die Anweisung 'PRINT'	
4.3.2.	Wahl des Ausgabeformat	
4.3.3.	Die Funktion 'TAB'	
4.3.4.	Die Anweisung 'DPL'	
4.3.5.	Die Anweisung 'CLEAR'	
4.3.6.	Graphische Darstellung	
4.4.	Programmverzweigungen	
4.4.1.	Vollständige strukturierte Verzweigung	
4.4.2.	Bedingte Abarbeitung einer Anweisung	
4.4.3.	Bedingter Sprung	
4.4.4.	Bedingte wiederholte Ausführung eines Programmblockes	
4.5.	Unbedingter Sprung	
4.6.	Laufanweisung	
4.7.	Unterprogrammaufruf	
4.8.	Kommentaranweisung 'REM'	
4.9.	Programmhalt und Programmunterbrechung	
4.10.	Programmende	
5.	Boolsche Ausdrücke und boolsche Wertzuweisung	V 1...2
6.	Standardfunktionen	VI 1...2
7.	Felder: Vektoren und Matrizen	VII 1...3
8.	Textverarbeitung	VIII 1...3

9.	Vereinbarungsteil, Einbindung von Assembler- funktionen	IX 1...17
9.1.	Lokaler Vereinbarungsteil	
9.2.	Globaler Vereinbarungsteil	
9.3.	Softwareschnittstellen der Assemblerprogramme	
9.4.	Aufruf von BASIC-Programmen aus Assembler- programmen	
10.	Prozeduren und Funktionen	X 1...5
11.	Weitere Beispiele	XI 1...4
12.	Syntaxdiagramme	XII 1...8
	Anhang	

1. Einleitung

Die vorliegende Beschreibung gibt eine detaillierte Übersicht über die Anwendung und Handhabung der Programmiersprache BASIC-80. Um die Erlernbarkeit der Sprache zu erleichtern, ist das Handbuch in zwei Teile gegliedert. Der erste Teil, die Abschnitte 1 bis 8 enthalten die notwendigen Informationen, um eigene Programme in BASIC-80 zu erstellen. Dem Anwender wird ein Umfang geboten, der in etwa dem Standard - BASIC entspricht. Die darauf folgenden Abschnitte stellen entscheidende Bestandteile von BASIC-80 vor, die Schnittstellen zu Prozeßgrößen ermöglichen und die Anwendung von Prozeduren und Anwenderfunktionen beschreiben. Daran schließt sich eine umfassende Beschreibung aller Anweisungen von BASIC-80 mit Syntaxdiagrammen an. Die Aufteilung der Beispiele in zwei Abschnitte hat zum Ziel, dem Anfänger und dem Fortgeschrittenen BASIC-80 nahezubringen.

BASIC-80 ist eine Programmiersprache für Mikroprozeßrechner. Im Sprachaufbau lehnt sich BASIC-80 an die bekannte Programmiersprache BASIC an. (Beginners All purpose Symbolic Instruction Code), die durch leichte Erlernbarkeit und Handhabung eine große internationale Verbreitung für Mikrorechner gefunden hat.

BASIC wurde Anfang der 60-er Jahre in den USA speziell für die Ausbildung in der Rechnerprogrammierung an Universitäten und Schulen entwickelt.

BASIC-80 (BASIC EXECUTION) wurde speziell für den Microcomputer MC 80 entwickelt und hat gegenüber dem Standard-BASIC Erweiterungen, die der Prozeßsteuerung dienen. Das sind beispielsweise Vereinbarungen, die es ermöglichen, Programme im Maschinencode des Mikroprozessors U 880 einzubeziehen. Damit können innerhalb eines BASIC-Programmes Prozeßgrößen angefordert oder ausgegeben werden. Die Sprache enthält Anweisungen, die boolesche Variable verarbeiten.

Mit BASIC-80 ist eine wesentliche Erweiterung des Leistungsumfanges des MC 80 gegeben. In Verbindung mit einem geeignetem Echtzeitbetriebssystem sind damit die meisten Probleme der Prozeßdatenverarbeitung und der Prozeßführung im Labor, bei der Entwicklung und Inbetriebnahme und in Rationalisierungsprojekten beherrschbar.

2. BASIC-80 - Grundlagen

BASIC-80 wurde auf der Basis des Prozessors U 880 realisiert. Die Programmlänge des Softwarepaketes beträgt 8 KByte. Davon sind 4 KByte Interpreter und 4 KByte Editor.

BASIC-80- Programme werden nicht in den Maschinencode übersetzt (keine Compilierung). Die Abarbeitung von BASIC-80 - Programmen erfolgt nach dem Interpreterprinzip. Der BASIC - Editor erzeugt aus BASIC - Befehlen einen speicherverschieblichen Zwischencode. Der Interpreter löst die Abarbeitung einer Folge von Maschinenbefehlen (U 880) aus, die vom Zwischencode bestimmt werden und das vom Anwender erstellte BASIC - Programm realisieren. Jedes BASIC - Programm ist nur mit dem Interpreter und dem Zwischencode lauffähig. Der Interpreter ist ein 4 KByte Programmpaket, das sich ab der Adresse 5000 bis 5FFF (hexadezimal) befinden muß und die Arithmetik und weitere wichtige Unterprogramme bereits mit enthält. Der Zwischencode mit dem Interpreter ist damit nach erfolgter Programm-erstellung im MC 80 auf jeder anderen Mikrorechnerkonfiguration mit dem Prozessor U 880 lauffähig.

Das weitere 4 KByte - Programmpaket BASIC - Editor dient mit dem Interpreter zur Erstellung und zum Test der BASIC - Programme. Der Editor ist nur im MC 80 lauffähig.

Der Zwischencode belegt weniger Speicherplatz als ein vergleichbares compiliertes Maschinencodeprogramm. Bei längeren Programmen (ab etwa 6 KByte) ergeben sich damit erhebliche Speicherplatzersparungen.

Das Abarbeitungsprinzip des Interpreters bedingt eine etwa 1,5 fache Laufzeit im Vergleich zu dem von einem Compiler erzeugtem Programm. Dieser Zeitfaktor ist bei der Problembearbeitung zu kalkulieren.

Die wesentlichen Bestandteile von BASIC-80 sind:

- Programmverzweigungen
- Ausgabeanweisungen
- vier Variablentypen
- Feldverarbeitung
- Vereinbarung von Assemblerfunktionen
- Gleitkommaarithmetik, Standardfunktionen.

Bei Programmverzweigungen werden bedingte Sprünge, Schleifen und Laufanweisungen unterschieden. Ausgabeanweisungen ermöglichen ein beliebiges Beschreiben des Bildschirms, auch mit einfachen graphischen Darstellungen.

Es gibt REAL-, INTEGER-, BOOLEAN- und STRING-Variable, die als lokale, globale oder formale Variable vereinbar sind. In BASIC-80 können Vektoren vereinbart und verarbeitet werden. Da eine beliebige Indizierung der einzelnen Elemente möglich ist, können Matrizenberechnungen ausgeführt werden.

Vereinbarungen von Assemblerfunktionen sind in BASIC-80 möglich und enthalten die Speicheradresse, ab der das gewünschte Assemblerprogramm steht.

Im folgenden ist ein einführendes Beispiel angegeben:

Die Funktion $c = \sqrt{a^2 + b^2}$ (Hypothense eines Dreiecks) soll im Tischrechnermodus berechnet werden. A und b sind vom Bediener einzugeben, anschließend wird c ausgedruckt usw.

- a) Die Anwahl von BASIC-80 erfolgt über die Menütabelle mit dreimaligem Betätigen von 'ENTER', vorher ist INIT auszuführen.
- b) Das folgende Programm läßt sich sofort eingeben, nach jeder Zeile ist ENTER zu betätigen.

```
20 PRINT 'A=';
  LET A=INP
  PRINT
  PRINT 'B=';
  LET B=INP
  PRINT
  PRINT 'HYPOTHENSE=';SQR(A*A+B*B)
  GOTO 20
```

Das Programm ist mit der Eingabe des Kommandos RUN und Betätigen von 'ENTER' zu starten, jede Zahleneingabe wird mit 'ENTER' abgeschlossen. Der Programmablauf kann mit der Taste 'OFF' unterbrochen werden, mit den Kursortasten '↑' und '↓' kann eine beliebige Programmstelle angewählt werden. Näheres dazu befindet sich im Abschnitt 3.

3. EDITOR BASIC-80

Der BASIC-Editor dient dem Erzeugen des Zwischencodes, dem Test erstellter Programme und der Korrektur (Überschreiben, Einfügen, Streichen) von Anweisungen.

3.1. Bedienkommandos

Anwahl

Die Auswahl von BASIC-80 erfolgt analog zu anderen Software-routinen des MC 80 über die Menütabelle. Nach Neueinschalten des Gerätes ist vorher INIT anzuwählen. Nach Aufruf meldet sich BASIC-80 mit der Ausschrift:

```
BASIC %C000 %E000
```

Diesen beiden hexadezimal angegebenen Adressen sind Vorzugsadressen, die den von BASIC verwendeten RAM-Bereich kennzeichnen. Sie können bei Bedarf geändert werden. Das erfolgt durch Überschreiben. Nach jeder Eingabe ist 'ENTER' zu betätigen. Bei der Anfangsadresse ist der niederwertige Teil immer 00 und kann nicht geändert werden.

Wird BASIC während des Programmierens oder-testens verlassen und erneut angewählt, wird der vorher gewählte RAM-Bereich wieder angeboten.

Nach erfolgter Auswahl befindet sich der Editor in der ersten Zeile in Eingabebereitschaft. Über dieser ersten Programmzeile befindet sich der in zwei Teile gegliederte Vereinbarungsteil. Der auf dem Display sichtbare lokale Vereinbarungsteil beginnt mit

```
DEF HAUPTPROGRAMM
```

Der darauf folgende lokale Vereinbarungsteil enthält die REAL - Vereinbarungen der Buchstaben A...Z. Das bedeutet, daß jedem Buchstaben eine Gleitkommazahl (real) zugeordnet werden kann. Jeder Buchstabe A...Z ist also Name einer REAL - Variable.

Der globale Vereinbarungsteil steht nicht sichtbar über dem lokalen Vereinbarungsteil am Anfang des gesamten Programmsystems. Über die Taste '↑' kann man ihn erreichen. Diese Vereinbarungen dürfen nicht unzulässig geändert werden.

Mit der Taste '↓' kann man sich im Programm wieder vorwärts bewegen, bis man am Ende den Eingabezustand wieder erreicht.

Bedienung, Bedienkommandos

Im Anzeigegrundzustand werden 6 Anweisungen angezeigt. Die 5 oberen Anweisungen sind einzeilig dargestellt, die letzte Anweisung wird vollständig ausgeschrieben. Eine Anweisung kann maximal drei Displayzeilen (95 Zeichen) umfassen:

Die unterste Anweisung, nach deren letzten Zeichen der Cursor steht, kann mittels Überschreiben und anschließendem Betätigen der ENTER-Taste korrigiert werden. Zum Überschreiben können alle Zeichentasten und die Tasten '←', '→' sowie 'CL' und die Tastenfunktionen cP und cL verwendet werden. Sie wirken repetierend.

Mit den beiden Tasten '↓' und '↑' kann jede Stelle im Programm ausgewählt werden. Jeweils die untere Zeile der Displayausschrift steht zur Korrektur bereit. Wird das Programmende erreicht, so wird automatisch auf den Eingabemodus umgeschaltet und weitere Anweisungen können hinzugefügt werden.

Im Eingabemodus ist das Display ab der 6. Zeile frei, die davorstehenden Anweisungen werden auf dem Display darüber einzeilig dargestellt. Es kann eine Anweisung oder ein Kommando eingetragen werden, der Abschluß (Übernahme) erfolgt mit 'ENTER'. Kommandos werden sofort ausgeführt, Anweisungen mit Zeilennummer werden in das Programm einsortiert und Anweisungen ohne Zeilennummer werden in die laufende Anweisungsfolge eingetragen. Die erste Anweisung eines Programmes muß mit einer Zeilennummer versehen werden. Bei der Eingabe von Anweisungen in das Programm wird dieser Eingabemodus solange beibehalten, bis eine andere Kommando-taste betätigt wird ('OFF', '↑', '↓', vgl. Tabelle).

Sollen in ein Programm Anweisungen eingefügt werden, wird nach Anwahl der Vorgängerzeile 'ENTER' betätigt. Im angenommenen Eingabemodus kann nun eine beliebige Anzahl von Anweisungen ergänzt werden.

Werden die Anweisungen mit Zeilennummern versehen, erfolgt ein automatisches Einsortieren und der Eingabemodus kann ab diesem Bereich weiter verwendet werden.

Alle zeichenerzeugende Tasten und die Tasten '←', '→' dienen der Texteingabe. Alle anderen Tasten '↓', '↑', 'OFF', 'ENTER', '⇨' sowie die Tasten der Control - Ebene sind Kommandotasten. Ihre Wirkung ist in der nachfolgenden Tabelle aufgeführt. Die Tasten der Control - Ebene sind erreichbar durch gleichzeitiges Betätigen der Controltaste (schwarze Taste mit Gefühlspunkt) und der angegebenen Zeichentaste. In der Tabelle sind diese Tastenfunktionen mit einem vorangestellten c gekennzeichnet.

Taste Aktivität

- ↓ Anwahl der nächsten Anweisung
- ↑ Anwahl der vorherstehenden Anweisung
- OFF Herstellen des Anzeigegrundzustandes
- cA \ Annahme des Eingabezustandes am Programmumfang
- cS Streichen der angewählten Anweisung
- cL Löschen des Zeichens auf der Cursorposition, der nachfolgende Text wird herangeschoben
- cP Erzeugen eines Leerzeichens auf der Cursorposition, der nachfolgende Text wird weggeschoben. Damit können im Text Einfügungen realisiert werden.
- ENTER Eingabezeile enthält Anweisung ohne Zeilennummer:
Die Anweisung wird in die laufende Anweisungsfolge eingetragen, auch Korrektur der Anweisung ist möglich.
- ENTER Eingabezeile enthält Anweisung mit Zeilennummer:
Die Anweisung wird entsprechend der Zeilennummer in das Programm einsortiert, die Anzeige erfolgt ab der neuen Position. Doppelte Zeilennummervereinbarungen werden angenommen. Einsortierung erfolgt nicht bei Korrektur.
- ENTER Eingabezeile enthält nur die Zeilennummer:
Anzeige ab dieser Zeilennummer
- ENTER Eingabezeile enthält ein Kommando laut nachfolgender Tabelle:
Ausführung dieses Kommandos

Taste Aktivität

ENTER	Zuvor wurde eine andere Kommandotaste ('↑', '↓', 'OFF') betätigt: Annahme des Eingabemodus; ermöglicht das Einfügen weiterer Anweisungen.
cN	Diese Funktion wird nur wirksam nach Programmunterbrechung mittels STOP - Anweisung oder OFF nach RUN! Ansonsten erfolgt die Fehlerausschrift >> DEF-EBENE. Schrittbetrieb oder Direktmodus, vgl. Abschnitt 3.3
cC	Programmfortsetzung nach STOP oder OFF nach RUN, vgl. 3.3

Tabelle 3.1 : Wirkung der Kommandotasten

Kommando	Aktivität
NEW	Löschen des gesamten von BASIC beanspruchten RAM-Bereiches, Treffen der Initialvereinbarungen
RUN	Start des geschriebenen Programms am Programmfang. Es können Fehlermeldungen auftreten, vgl. 3.2 Das laufende Programm kann mit 'OFF' unterbrochen werden. Dabei wird der Anzeigegrundzustand an der Unterbrechungsstelle angenommen.
BYE	Verlassen des BASIC - Systems
PUT name	Auslagern des geschriebenen Programms
PUT name Zeilennummer	Auslagern bis ausschließlich Zeilennummer
PUTD	wie PUT Löschen des Quellbereiches
GET name	Einfügen eines vorher ausgelagerten Programmstückes
LIST	Ausdrucken des Gesamtprogrammes auf einem Drucker

Tabelle 3.2 : Kommandos nach 'ENTER' und deren ausgelöste Aktivität

Alle Kommandos nur im Eingabemodus
(freie Zeile, ggf. vorher ENTER)
eintragen!

Anmerkung zum Kommando LIST

Voraussetzung für die Verwendung ist eine Druckerinitialisierung und eine Druckeroutine :

- Auf der Adresse 27FA steht ein Sprung zur Druckerinitialisierung (DRI)
- Auf der Adresse 27FD steht ein Sprung zur Druckeroutine (DRU)
- die Unterprogramme werden mit RETURN abgeschlossen

Die Angabe der Pufferlänge erfolgt im Akkumulator und das erste zu druckende Zeichen ist Inhalt von Register DE.

Ein Zeilenende wird mit ØA und ØD gekennzeichnet. Das Kommando LIST ist nach der ersten Eingabezeile einzugeben.

Arbeit mit PUT und GET

Das Kommando PUT hat zwei Funktionen. Es dient einmal dem Auslagern des Gesamtprogramms, das dann als Assemblerfunktion verwendet werden kann.

Unter Angabe einer Zeilennummer kann mit PUT oder PUTD ein Programmteil ausgelagert werden, der mit GET an anderer Stelle wieder eingefügt werden kann. Damit ist eine Umsortierung der Programmanweisungen möglich. Zu Beachten ist, daß dabei die Zeilennummerordnung verletzt werden kann.

Die Kommandos PUT, PUTD sowie GET beziehen sich auf einen Pufferbereich, der als Assemblerfunktion deklariert werden muß. Am Ende des globalen Vereinbarungsteiles muß deshalb hinzugefügt werden (Bsp.) :

```
GLOBAL %E000  
DCL PUF
```

Damit wird PUF als Assembleranweisung vereinbart. Im Anwendungsfall wird damit jedoch ein Pufferbereich für PUT und GET geschaffen, der ab der Adresse E000 (hexadezimal) beginnt.

Im folgenden Beispiel werden die Kommandos PUTD und GET verwendet :

```
GLOBAL %E000  
DCL PUF  
DEF HAUPTPROGRAMM  
REAL A,B,C,D,E,F,G,H  
REAL I,J,K,L,M,N,O,P,Q  
REAL R,S,T,U,V,W,X,Y,Z  
10 FOR A=1 TO 8  
PRINT A,SQR(A)  
NEXT A  
20 LET B=1  
LET C=INP  
FOR D=B TO C  
LET B=B+1/C  
NEXT D  
30 END
```

} am Ende des globalen Vereinbarungsteiles anfügen (↑ und ENTER)

} sichtbarer lokaler Vereinbarungsteil

Die Zeile 10 (↑) wird angewählt. Danach wird das Kommando

```
PUTD PUF 20
```

einggegeben und mit 'ENTER' abgeschlossen. Es werden die Anweisungen ab der angewählten Programmzeile bis ausschließlich der Zeile mit der Nummer 20 in den Pufferbereich übernommen und aus dem Programm gestrichen.

Danach wird die Zeile NEXT D angewählt, 'ENTER' betätigt und das Kommando GET PUF ausgeführt.

ausgeführt. Damit wird der Pufferinhalt nach der Anweisung NEXT D in das Programm eingefügt und es ergibt sich folgende Anweisungsreihenfolge :

```
20 LET B=1
   LET C=INP
   FOR D=B TO C
   LET B=B+1/C
   NEXT D
10 FOR A=1 TO 8
   PRINT A,SQR(A)
   NEXT A
30 END
```

Man beachte die falsche Zeilennummerreihenfolge. Diese kann gegebenenfalls korrigiert werden.

Um in den Pufferbereich einschreiben zu können, muß er leer sein. Ansonsten erfolgt die Fehlermeldung " » SPEICHER VOLL ".

Das Kommando GET löscht den Pufferbereich, so daß er wieder für das Kommando PUT frei ist.

Mit dem Kommando GET läßt sich ein BASIC-Programm von einem Speicherbereich außerhalb des BASIC-Systems holen. Danach läßt es sich bearbeiten (korrigieren, testen).

Mit PUT ist es dann wieder auf den Originalbereich auszulagern und kann dann als eigenständiges Assemblerprogramm (Task) wieder gestartet werden.

Beim erstmaligen Studium dieses Handbuches empfiehlt es sich beim Abschnitt 4 fortzusetzen und auf die beiden folgenden Punkte bei Bedarf zurückzugreifen.

3.2. Fehlerausschriften

Syntaxfehler bei der Eingabe :

Nach Korrektur oder Neueingabe einer Anweisung und Betätigen von 'ENTER' erfolgt die Übersetzung der Zeile. Dabei wird ein Teilsyntaxtest durchgeführt. Jedes einzelne Wort wird in der Übersetzungstabelle und im Vereinbarungsteil gesucht. Wird es nicht gefunden, so wird der Cursor unter die unbekannte Zeichenfolge gesetzt. Es kann sofort korrigiert werden. Danach wird wieder 'ENTER' betätigt.

Eine Überprüfung der Reihenfolge der Einzelworte erfolgt bei der Eingabe nicht. Um den gesamten Syntaxtest des Programms auszuführen, ist die Abarbeitung des Gesamtprogramms notwendig.

>> SYNTAX

Diese Ausschrift erfolgt nach RUN, wenn in einem Programm die Anordnung der Einzelworte einer Anweisung fehlerhaft ist und diese Anweisung abgearbeitet wird. Die Programmabarbeitung wird unterbrochen, es kann nur mittels RUN neu gestartet werden. Die fehlerhafte Stelle wird im Anzeigegrundzustand mit >> gekennzeichnet. Es ist zu beachten, daß bei der Abarbeitung z. T. nicht die eigentliche Fehlerstelle, sondern ein Folgefehler erkannt wird. Der eigentliche Fehler kann sich auch davor befinden. Nach Erscheinen dieser Fehleranzeige muß die Taste 'OFF' betätigt werden. Danach kann die betreffende Anweisung korrigiert werden (Überschreiben, ENTER). Das Programm muß mit RUN neu gestartet werden.

>> ZUWEISUNG

Diese Ausschrift erfolgt nach RUN, wenn in einem Programm die Anweisungsklammern fehlerhaft gesetzt sind.

Mit Schachtelungsmöglichkeiten muß folgen auf :

IF ... DO ELSEDO ... DOEND

IF ... DO' DOEND

IF ... THEN....Zeilennummer mit Anweisung

FOR...TO NEXT

WHILE ... DO LOOP

GOTO Zeilennummer Zeilennummer

Fehlt ein Anweisungsklammerabschluß (DOEND, LOOP, NEXT, Zeilennummer), so erfolgt die Fehleranzeige am Programmende bzw. bei einer END - Anweisung. Ist ein Anweisungsklammerabschluß zuviel vorhanden, so wird dieser als fehlerhaft angezeigt.

Nach Fehlerkorrektur und erneutem Programmstart mit RUN, kann ein weiterer Fehler dieser Art angezeigt werden usw. Erst nach Korrektur aller Anweisungsklammerfehler wird das Programm nach RUN gestartet.

» SPEICHER VOLL

Diese Aussschrift erfolgt, wenn eine Anweisung eingegeben werden soll und der Programmspeicherbereich würde dabei überfüllt werden. Der vom BASIC- Programm insgesamt verfügbare Speicherbereich wird zu Beginn beim Aufruf (Abschnitt 3 .1., Anwahl) angegeben. Bei der Programmabarbeitung wird von oben (von der Endadresse aus rückwärts) der Variablenspeicher in Form eines arithmetischen Stacks angelegt. Von unten her wird das Programm eingeschrieben. Eine Fehleranzeige des Programmspeicherüberlaufs erfolgt, ein möglicher Überlauf des Variablenspeichers wird nicht kontrolliert. Das ist bei umfangreichen Programmen mit mehreren geschachtelten Prozeduraufrufen zu beachten.

» DEF-EBENE

Diese Fehleranzeige erfolgt beim Auslösen der Kommandos cN oder cC, wenn das Programm nicht im Lauf unterbrochen wurde oder eine falsche Prozedurebene angewählt wird.

3.3. Programmtest

Die interaktive und interpretative Arbeitsweise von BASIC-80 ermöglicht einen umfassenden Programmtest zur Feststellung der pragmatischen Fehlerfreiheit. Ein Programm kann in Teilen geschrieben und getestet werden. In jedem Programmzweig können zur Kontrolle PRINT- und PAUSE - Anweisungen eingefügt werden, die Zwischenergebnisse ausdrucken. Damit kann die Datenentwicklung beobachtet werden.

Im Beispiel des Abschnitt 7 : Stürzen einer Matrix soll die Richtigkeit der Zuordnung der Indizes in der Schleife beobachtet werden.

Nach der Anweisung

```
LET MAT2(Jx3+I)=MAT1(Ix3+J)
```

kann man hinzufügen:

```
PRINT 'MAT1(';Jx3;' ,';I;')=MAT2(';Ix3;' ,';J;')
```

```
WAIT
```

Damit wird in jeder Schleife der Ausdruck der aktuellen Indizes in der Form (Bsp.)

```
MAT1(2,3)=MAT2(3,2)
```

realisiert.

Da das Druckformat wahrscheinlich ungünstig ist, kann man am Programmanfang zufügen :

```
5 PRINT FORMAT 0.3
```

Die zusätzlich eingefügten Anweisungen können nach folgendem Test wieder gestrichen werden.

Die Anweisung STOP und die Kommandos cN und cC

Mit der Anweisung STOP kann ein Programm an beliebiger Stelle angehalten werden. Um den Bildschirminhalt im Moment der Abarbeitung der STOP- Anweisung nicht zu zerstören, wird auf dem Display rechts unten nur STOP vermerkt. Erst nach Betätigen einer beliebigen Taste wird der Editormodus eingenommen. Voraussetzung für eine erfolgreiche Weiterarbeit ist, daß das Kommando RUN in dieser Prozedurebene (im allgemeinen im Hauptprogramm) eingegeben wurde.

Durch einfaches Betätigen von 'ENTER' kann nun wie gewohnt der Eingabemodus angenommen werden. Wenn man als Anweisung

!A

eingibt und nicht die ENTER- Taste, sondern die Taste cN betätigt (Control - Taste und Taste N gleichzeitig), so wird diese Anweisung nicht in das Programm eingetragen, sondern sofort ausgeführt. Als Ergebnis der Ausführung erscheint auf dem Display der Wert der Variablen A. Diese Ausführung der Anweisung wird auch als Direktmodus bezeichnet. Die Anweisung

!variablenname

dient also dem Ausschreiben eines Variablenwertes ähnlich der PRINT - Anweisung.

Als weitere Anweisung im Direktmodus läßt sich beispielsweise eingeben :

A=5 cN

Damit wird die Anweisung LET A=5 sofort ausgeführt und die Variable erhält den Wert 5.

Auch jede beliebige andere Anweisung läßt sich so ausführen, was aber nicht in jedem Falle sinnvoll ist.

Wird z. B. geschrieben :

!3*4 cN

so wird der Wert 12 ausgedruckt. Damit läßt sich der Direktmodus für Taschenrechnerzwecke nutzen.

Wird jetzt die Taste 'OFF' betätigt, so wird der Anzeigegrundzustand wieder eingenommen (vgl. Kommandotabelle nach Abschnitt 3.1).

Ein Betätigen von

cN ohne Eingabe einer Anweisung

löst die Ausführung der im Programm stehenden Anweisungen aus. Damit wird eine Abarbeitung des Programms im Einzelschritt erreicht. Prozeduraufrufe werden im Komplex abgearbeitet.

Gibt man das Kommando

cC

ein, so wird das Programm zur weiteren Abarbeitung an diesem Punkt gestartet.

Das vorige Anwählen einer anderen Programmposition mittels der Tasten '↓' und '↑' ist möglich. Jedoch darf die Prozedurebene nicht verlassen werden. In einer anderen Prozedurebene gelten andere lokale Variable, die nur über die Programmabarbeitung erreicht werden können. Deshalb ist das Verlassen der Prozedurebene nicht sinnvoll.

Soll eine Prozedur selbst im Schritt getestet werden, so ist eine STOP - Anweisung in die Prozedur einzufügen. Das RUN - Kommando ist bei Anzeige der Prozedur einzugeben und die Prozedur ist im Hauptprogramm mit den nötigen aktuellen Variablen und Parametern aufzurufen (vgl. dazu Abschnitt 10).

4. Anweisungen

Im Standard - BASIC beginnt jede Anweisungszeile mit einer Zeilennummer, die Anweisungen sind in steigender Reihenfolge der Zeilennummern sortiert. Bei BASIC-80 muß nicht jede Zeile numeriert werden. Zeilennummern sind notwendig:

- Bei der ersten Anweisung des Programms (aber nicht im Vereinbarungsteil),
- nach einer Kommentaranweisung (REM),
- nach IF ... THEN ... als Abschluß der Aktivität,
- als Sprungziel (GOTO).

Eine numerierte Zeile mit den nachfolgenden Zeilen ohne Nummer entspricht in anderen BASIC - Versionen einer komplexen Programmzeile mit mehreren Anweisungen.

Sämtliche vorgestellten Anweisungen können sofort am MC 80 notiert, getestet und diskutiert werden. Der Anfänger sollte sich dabei streng an die Reihenfolge der Vorstellungen halten. Der Fortgeschrittene kann weiterführende und Sonderfälle zusätzlich testen. Bei Fehleranzeigen (Syntaxfehler u. ä.) ist der Abschnitt 3 zu Rate zu ziehen.

Zu Beginn ist am MC 80 die Funktion INIT auszuführen und BASIC-80 anzuwählen (Menütabelle).

Danach lassen sich wie im ersten Beispiel die Programme eingeben und starten. Das Kommando NEW mit Abschluß durch 'ENTER' bewirkt das Löschen aller Eintragungen und Wiederherstellung des Initialzustandes.

4.1. Wertzuweisung 'LET'

BASIC-80 verfügt als Initialvereinbarung über Variable mit den Bezeichnungen A,B,C,...,Z. Mit der Anweisung LET können diesen Variablen Zahlenwerte zugewiesen werden, die direkt angegeben werden oder Ergebnisse eines arithmetischen Ausdruckes sind. Als Beispiel kann notiert werden :

```
10 LET A=5
   LET B=A*A+3
   LET C=B/A*SIN(B)
   LET X=4
```

Arithmetische Ausdrücke werden wie in der Mathematik gewohnt notiert, Punktrechnung geht vor Strichrechnung, Klammern werden mit Vorzugsrecht bearbeitet, mehrfache Klammerung ist möglich. Das Multiplikationszeichen wird mit '*' und das Divisionszeichen mit '/' dargestellt. Exponieren ist nicht direkt möglich. Mehrzeilige Bruchdarstellung ist als einzellige Form mit Klammern aufzulösen. Im Beispiel entspricht der Anweisung

LET C=B/A*SIN(B) der arithmetische Ausdruck

$$c = \frac{b}{a} \cdot \sin b$$

Der arithmetische Ausdruck

$$d = \frac{a + b}{a - b} \quad \text{muß notiert werden :}$$

LET D=(A+B)/(A-B)

Die Argumente der Funktionen sind stets in Klammern () zu setzen.

Die Standardfunktionen sind im Abschnitt 6 beschrieben. Funktionen können auch selbst definiert werden, siehe Abschnitt 9 und 10.

Nach Eingabe der Beispielsanweisungen kann das Programm mit dem Kommando RUN und Betätigen von 'ENTER' gestartet werden. Nach fehlerfreier Abarbeitung erfolgt auf dem Display rechts unten die Meldung READY. Nach Betätigen einer beliebigen Taste steht das Programm wieder im Editormodus am Programmanfang. Über die Kursortaste \downarrow kann man sich an das Programmende bewegen. Im Beispiel wurden bei der Programmabarbeitung den Variablen die Werte zugewiesen. Die Zuweisung erfolgt im Programmablauf nach außen hin nicht sichtbar. Die Variablenwerte kann man erst durch eine Ausgabeanweisung erfahren (z. B. PRINT). Wir schreiben deshalb am Programmende hinzu :

```
PRINT A
PRINT B
PRINT C
PRINT X
```

Nach erneutem Start (RUN und ENTER) stehen auf dem Display untereinander die vier zugewiesenen Zahlenwerte.

Bei jedem Programmstart mittels RUN bleiben entgegen manchen anderen BASIC - Versionen vorher zugewiesene Variablenwerte erhalten. Wird eine Variable in einem arithmetischen Ausdruck verwendet ohne ihr vorher einen Wert zuzuordnen, so enthält sie den Wert der größten darstellbaren negativen Zahl -170.10^{36}

Eine Fehlermeldung erfolgt nicht. Wenn im Vereinbarungsteil korrigiert wird (vgl. Abschnitt 9), können sich Veränderungen der Zuordnung der Werte zu den Variablen ergeben, was beim Neustart zu berücksichtigen ist.

Der Ausdruck nach LET ist nicht als mathematische Gleichung, sondern als Zuweisungsoperation zu verstehen. In anderen höheren Programmiersprachen wird dafür das Zeichen '=' verwendet. Im Unterschied dazu wird bei Bedingungsabfragen das Zeichen '=' auch als Vergleichsoperator verwendet.

Nach Zeilennummern kann das Schlüsselwort LET auch entfallen.
Nach dem Schlüsselwort DO kann ebenfalls direkt ohne LET eine Wertzuweisungsoperation notiert werden.

4.2. Die Funktion 'INP'

Mit der Funktion INP können Texte oder Zahlenwerte über die Tastatur eingegeben werden. Um der Variablen A einen Wert über die Tastatur zuordnen zu können, kann man notieren :

```
10 LET A=INP
```

Die Programmabarbeitung verharret an dieser Stelle, erwartet auf der aktuellen Cursorposition eine Eingabe. Die Übernahme und Programmfortsetzung erfolgt durch Betätigen von 'ENTER'.

Im Beispiel steht nach RUN der Cursor in der unteren Zeile am Anfang. Zwecks besserer Bedienerführung ist es günstiger, vorher einen Begleittext auszuschreiben. :

```
10 PRINT 'A=';  
LET A=INP
```

In diesem Beispiel wird auf der letzten Zeile zuerst "A=" ausgeschrieben. Das Semikolon am Ende der PRINT-Anweisung verhindert die Zeilenschaltung. Die Eingabe wird nach der Ausschrift "A=" erwartet. Die nächste PRINT-Anweisung wird dann ab der neuen, durch die Eingabe veränderten Cursorposition ausgeführt. Eine leere PRINT-Anweisung

```
PRINT
```

bewirkt eine einfache Zeilenschaltung.

Werden Zahlen erwartet, so wird bei keiner oder falscher Eingabe eine 0 übernommen. Die Zahlen müssen mit dem Vorzeichen oder einer Ziffer beginnen. INP kann auch zur Eingabe von Texten benutzt werden. Das ist im Abschnitt 8 beschrieben.

4.3. Ausgabeanweisungen

Ausgabeanweisungen sind ausschließlich auf den MC 80 - Bildschirm bezogen. Wenn andere Bildschirmgeräte, Drucker oder andere Datenspeicher angesteuert werden sollen, müssen entsprechende Treiber-routinen als Assemblerprogramm ergänzt und im Vereinbarungsteil vermerkt werden (vgl. Abschnitt 9).

4.3.1. Die Anweisung 'PRINT'

Die Anweisung PRINT wurde in den vorangegangenen Beispielen bereits mehrfach benutzt, um den Wert von Variablen auf dem Display erscheinen zu lassen. Mit dieser Anweisung kann jeweils genau eine Displayzeile beschrieben werden. Das ist im Normalfall die unterste Displayzeile. Danach wird der gesamte Displayinhalt um eine Zeile nach oben verschoben.

Die Anweisung PRINT kann zum Ausdrucken von Zahlenwerten und Begleittexten verwendet werden. Im Beispiel wird auf dem Display die Ausschrift

B=15

erzeugt :

```
10 LET B=10+5
   PRINT 'B=';B
```

Werden in einer PRINT-Anweisung eine Variable oder ein arithmetischer Ausdruck angegeben, so erscheint auf dem Display der zugehörige Zahlenwert. Auszudruckende Begleittexte werden in '...' geklammert. Das Semikolon ist ein einfaches Trennzeichen. Wird statt dessen als Trennzeichen ein Komma verwendet, so wird ein Leerzeichen zwecks Tabulierung vorgerückt, mehrere Komma sind möglich. Damit können günstig Tabellen gestaltet werden.

Ein Semikolon am Ende der PRINT-Anweisung verhindert die Zeilenschaltung, es kann auf der gleichen Zeile weiter ausgegeben werden.

Mit der PRINT-Anweisung ist es möglich, den Inhalt von Textvariablen wiederzugeben. Dieses ist detailliert im Abschnitt 8 beschrieben.

4.3.2. Wahl des Ausgabeformatates

Bisher wurde das Ausgabeformat von Zahlen nicht beeinflußt. BASIC-80 enthält eine Formatsteuerung die es gestattet, die Stellenzahl vor und nach dem Komma festzulegen.

Wird das Format nicht festgelegt, werden insgesamt 6 Stellen ausgeschrieben. Die Anweisung PRINT 7 bewirkt dann folgende Displayausschrift:

7.00000

Zahlen größergleich 1000 werden mit Exponent dargestellt, der in Dreierschritten gezählt wird, z.B.:

1.23000E+03

Zahlen von 1 bis kleiner als 1000 werden immer ohne Exponent dargestellt.

Das Zahlenformat kann geändert werden, indem die PRINT-Anweisung mit einer Formatsteuerung ergänzt wird.

Die Formatanweisung hat folgende Struktur:

PRINT FORMAT !a.b! — tabellengerecht
 ohne !: textgerecht
 Nachkommastellen 0...7
 Vorkommastellen 1...7
 0: Ingenieurformat
 Exponent wird nicht dargestellt, Überlaufanzeige; ohne !: Exponent wird eingeplant

- Ingenieurformat:

Es wird eine Mantisse im Bereich 1...999.9 und ein Exponent in Dreierschritten dargestellt, z.B.

1.5
123
12.45E+03
1.267E-12

Die Nachkommastellenzahl entspricht hier der Gesamtstellenzahl, sie muß ≥ 3 sein, der Dezimalpunkt wird entsprechend der Vorkommastellenzahl (1...3) gerückt.

- **Kein Ingenieurformat:**
Die Vorkommastellenzahl beträgt 1 bis 7. Ist die Zahl zu groß (größere Vorkommastellenzahl nötig), so wird ein Exponent berechnet und ggf. ausgegeben oder es erfolgt eine Überlaufanzeige. Ist die Zahl kleiner als mit der gewünschten Stellenzahl darstellbar, so wird 'Ø' ausgeschrieben (keine Berechnung von negativen Exponenten).
- **Tabellengerecht:**
Die Zahlen werden so ausgeschrieben, daß unabhängig vom Wert die gleichen Druckpositionen belegt werden:
 - . Bei Vorzeichen '+' wird ein Leerzeichen ausgegeben
 - . führende Nullen werden mit Leerzeichen dargestellt
 - . nachfolgende Nullen werden ausgeschrieben
 - . außer bei Wahl des Ingenieurformates steht der Dezimalpunkt an der gleichen Stelle
 tritt kein Exponent auf, kann er aber eingeplant werden (vorderes! nicht gesetzt), indem 4 Leerzeichen dargestellt werden. Nach dem Exponenten wird ein nachfolgendes Leerzeichen angegeben.
- **Textgerecht:**
Alle führenden und folgenden Informationen ohne Informationsgehalt werden weggelassen: Vorzeichen '+', führende Nullen, nachfolgende Nullen, Exponent.
- **Überlaufanzeige:**
Im Druckbereich wird kein Exponent vorgesehen. Überschreitet die Zahl den möglichen Anzeigebereich, so wird auf der letzten Druckposition ein 'Ø' als Fehlerprotokollierung geschrieben. Ist das vordere '!' nicht gesetzt, so werden bei tabellengerechter Darstellung 5 Leerzeichen bei fehlendem Exponent ausgegeben. Die Überlaufanzeige (Angabe des vorderen !) ist semantisch sinnlos bei Ingenieurformat und teilweise bei textgerechter Darstellung (vgl. Beispiele).
- **Rundung:**
Es wird auf die vorgegebene Stellenzahl auf- bzw. abgerundet
- **Maximale Stellenzahl:**
Es werden niemals mehr als 7 Stellen ausgegeben.
- **Standardformat:**
Wird keine Formatanweisung angegeben, so ist das FORMAT Ø.6! festgelegt.

Eine einmal getroffene Formatvereinbarung wird solange weiterverwendet, bis eine andere Vereinbarung des Formates erfolgt.

Beispiele:

Zahl:	1.5	0.078	1024	-15
FORMAT Ø.4	1.5	78E-Ø3	1.Ø24E+Ø3	-15
FORMAT Ø.4!	1.5ØØØØØØØØ	78.ØØE-Ø3	1.Ø24E+Ø3	-15.ØØØØØØØØ
FORMAT 2.2	1.5	Ø.Ø8	1Ø.24E+Ø2	-15
FORMAT 2.2!	1.5ØØØØØØØØ	Ø.Ø8ØØØØØØØØ	1Ø.24E+Ø2	-15.ØØØØØØØØ
FORMAT !2.2!	1.5Ø	Ø.Ø8	1Ø.2Ø	-15ØØ

4.3.3. Die Funktion 'TAB'

TAB ist eine Funktion mit einem Argument. Entsprechend dem Ganzzahlanteil des Arguments werden Leerzeichen ausgeschrieben.

TAB kann Bestandteil der Ausgabeanweisung PRINT und DPL sein.

Im Beispiel :

```
10 LET A=5  
   PRINT '+';TAB(A);'+'
```

werden zwischen den beiden Zeichen '+' 5 Leerzeichen ausgegeben.

Im folgenden Beispiel wird eine Sinushalbwelle mit '*' ausgeschrieben:

```
10 FOR A=0 TO PI STEP PI/8  
   PRINT TAB(SIN(A)*20);'*'  
   NEXT A
```

4.3.4. Die Anweisung 'DPL'

Mit der Anweisung 'DPL' kann das Display an beliebiger Stelle beschrieben werden. Die Argumente der DPL - Anweisung und die Bildschirmausschrift selbst entspricht formmäßig der PRINT- Anweisung. Es erfolgt kein Bildrollen. Nach DPL werden die Zeile und die Spalte angegeben, ab der gedruckt werden soll. Bsp.:

```
10 DPL 4,5 'A=';A  
    | |  
    | L---Spalte  
    | L-----Zelle
```

Die Zählung von Spalte und Zeile beginnt mit 0. Anstelle der Zahlen können auch arithmetische Ausdrücke angegeben werden.

Beispiel:

```
10 FOR A=0 TO 7  
   DPL A,0 '*'  
   NEXT A  
   FOR A=0 TO 31  
   DPL 7.75-A/4,A '*'  
   NEXT A
```

Bildausschrift

```
*               *****  
*               *****  
*               *****  
*               *****  
*               *****  
*               *****  
*               *****  
*               *****  
*               *****
```

4.3.5. Die Anweisung 'CLEAR'

Die Anweisung CLEAR bewirkt ein Löschen des Displays. Nach RUN wird das Display automatisch gelöscht.

4.3.6. Graphische Darstellung

Im BASIC-80 besteht die Möglichkeit, eine analoge Kurve darzustellen. Dafür entfällt die zweite Displayzeile. Die graphische Darstellung wird mit der Anweisung

GRAPH

eingeschaltet und mit der Anweisung

GRAPH AUS

abgeschaltet. Die Zuweisung von Werten erfolgt über die Ausgabe-funktion GRA. Mit jeder Wertzuweisung zu GRA wird, auf der Kurve rechts dieser Wert zugefügt und die Kurve nach links verschoben. Die Abszisse hat 253 Werte, die Ordinate umfaßt den Wertebereich von

$$-128 \leq \text{GRA} \leq 127.$$

Werden größere Werte angeboten, erfolgt keine Begrenzung. Im Beispiel wird eine exponentiell abklingende Sinusschwingung dargestellt :


```
10 GRAPH
   FOR A=0 TO 25.3 STEP 0.1
   LET GRA=100*SIN(A)*EXP(-A/20)
   NEXT A
```

Bei Verlassen der Programmbearbeitung bleibt die Graphik eingeschaltet, wenn die Anweisung GRAPH AUS nicht abgearbeitet wurde.

4.4. Programmverzweigungen

4.4.1. Vollständige strukturierte Verzweigung


Programmzweige müssen oft in Abhängigkeit bestimmter Bedingungen durchlaufen werden. In einem Programmablaufplan (PAP) wird eine vollständige strukturierte Programmverzweigung folgendermaßen dargestellt :



Jeder Zweig kann selbst wieder beliebig in sich strukturiert sein, also weitere Verzweigungen enthalten, die wieder zusammengeführt werden.

Eine Verzweigung könnte z. B. folgendermaßen aussehen :

```
10 LET A=5
   LET X=4
   IF A>X DO
   LET Y=1
   ELSEDO
   LET Y=2
   DOEND
   PRINT Y
```



als BASIC-Programm

und als PAP.

Die Bedingung befindet sich zwischen IF und DO.

Der Ja - Zweig wird von DO und ELSEDO eingeschlossen, der Nein-Zweig befindet sich zwischen ELSEDO und DOEND. Die PRINT - Anweisung dient der Kontrolle der Programmausführung und stellt die Programmfortsetzung nach der Verzweigung dar.

Als Vergleichsoperatoren sind zugelassen :

- > größer
- < kleiner
- >= größergleich
- <= kleinergleich
- = gleich
- <> ungleich

Als Bedingung können aber auch beliebige boolesche Ausdrücke verwendet werden. Dazu sei auf die Syntaxdiagramme und spätere Beispiele verwiesen.

Der NEIN-Zweig dieser vollständigen Programmverzweigung kann entfallen, indem der JA - Zweig sofort mit DOEND abgeschlossen wird. ELSEDO tritt dann nicht auf. Beispiel : Betragsbildung

```
IF A<0 DO
LET A=0-A
DOEND
```


Soll der Ja- Zweig entfallen, können die Anweisungen zwischen DO und ELSEDO weggelassen werden. Meist läßt sich die Bedingung dann aber anders formulieren. Bsp: Erhalten des negativen Betrages

```
IF A<0 DO
ELSEDO
LET A=0-A
DOEND
```

4.4.2. Bedingte Abarbeitung einer Anweisung

Eine weitere Formulierungsmöglichkeit der Programmverzweigung ist die IF ... THEN Anweisung. Diese Formulierung ist aus Kompatibilitätsgründen mit anderen BASIC-Versionen in BASIC-80 aufgenommen worden. Sie ermöglicht aber nur die Realisierung des Ja-Zweiges ohne weitere Strukturierungsmöglichkeit. Der Abschluß des Ja-Zweiges wird durch Verwendung einer Zeilennummer bei der nächsten Anweisung im Programm gekennzeichnet. Dazu folgendes Beispiel :

```
IF A>X THEN
LET Y=2
20 PRINT Y
```




als BASIC-Programm und als PAP.

4.4.3. Bedingter Sprung

Der bedingte Sprung wird mit der IF ... THEN Anweisung realisiert. In Abhängigkeit von einer Bedingung kann der Programmlauf ab einem beliebigen Punkt fortgesetzt werden.

Beispiel :

```
IF A > X THEN
GOTO 100
40 LET Y=0
PRINT Y
END
.
.
.
100 LET Y=10
PRINT Y
.
.
```



als BASIC - Programm und als PAP.

Ist die Bedingung (A > X) erfüllt, so wird das Programm ab der Zeile 100 fortgesetzt.

Die Möglichkeit birgt die Gefahr der nichtstrukturierten Programmierung. Sie sollte deshalb möglichst vermieden werden.

Eine nicht strukturierte Programmierung ist vom ersten Anschein oft optimaler, zeit- und speicherplatzsparender. Bei Programmveränderungen und -erweiterungen besteht die Gefahr, dass eine Vielzahl von GOTO-Befehlen die Übersicht erschwert. Man spricht von sogenannter "Spagettiprogrammierung". In jedem Falle sollte versucht werden, strukturiert zu programmieren.

Nur bei Aussprüngen aus dem Gesamtalgorithmus, z. B. bei Fehlermeldungen und Beenden der Programmbearbeitung kann ein solcher bedingter Sprung nützlich sein.


4.4.4. Bedingte wiederholte Ausführung eines Programmblockes
 Mit der WHILE - Anweisung kann eine Schleife beliebig oft in Abhängigkeit von einer Bedingung durchlaufen werden. Dabei ist zu garantieren, daß sich die Bedingung während des Schleifendurchlaufes ändert, andernfalls verharrt das Programm an dieser Stelle in einer Eigenschleife. Der Abschluß der zur Schleife gehörenden Aktivitäten wird mit LOOP gekennzeichnet.

Als Beispiel sei ein Programm zur Berechnung einer Zahlenfolge mit Abbruchkriterium angegeben.

Hinweis: NEW und ENTER bewirken das Herstellen des Initialzustandes

```

10 LET A=1
   LET B=1
   LET E=1
   WHILE A > 2E-06 DO
   LET A=A/B
   LET B=B+1
   LET E=E+A
   LOOP
   PRINT E
  
```



als BASIC-Programm

und als PAP .

Das Ergebnis des Programms ist die Zahl e (Basis des natürlichen Logarithmus) nach der Zahlenfolge


$$e = \lim_{n \rightarrow \infty} \left(1 + \frac{1}{1!} + \dots + \frac{1}{n!} \right)$$

Die Bedingung ist hier das Abbruchkriterium $A > 0.000002$. Die Bedingung steht zwischen WHILE und DO, wie bei IF...DO können auch beliebige boolesche Ausdrücke verwendet werden. Die Anweisungsfolge, die bei erfüllter Bedingung abzuarbeiten ist, steht zwischen DO und LOOP. Bei nicht erfüllter Bedingung wird nach LOOP fortgesetzt.

Im Sinne der wissenschaftlich technischen Fortschritts ergibt sich die folgende Ergänzung, die sich auf die bedingte wiederholte Ausführung eines Programmblockes bezieht.


Mit der folgenden Konstruktion ist eine Programmschleife realisierbar, welche die Abbruchbedingung in der Schleife abfragt:

```
BEGIN LOOP  
anweisungen 1  
IF bedingung DO  
anweisungen 2.  
LOOP
```



Damit läßt sich auch die Austrittsbedingungsabfrage am Schleifenende realisieren:

```
BEGIN LOOP  
anweisungen  
IF bedingung DO  
LOOP
```



Ein weiteres Beispiel stellt eine Zeltschleife dar :

```
LET A=500  
WHILE A>0 DO  
LET A=A-1  
LOOP  
PRINT A
```

Mit diesem Beispiel wird eine Zeltschleife von ca. 4 Sekunden realisiert.

4.5. Unbedingter Sprung

Wenn das Programm unbedingt oder in einem bedingten Zweig an anderer Stelle fortgesetzt werden soll, kann die Anweisung (Bsp.)

```
GOTO 100
```

benutzt werden. Das Programm wird an der mit der Zeilennummer 100 markierten Zeile fortgesetzt. Diese Anweisung wurde bereits im Abschnitt 4.4.3. verwendet. Es gelten die dort ausgeführten Bemerkungen bezüglich strukturierter Programmierung.

4.6. Laufanweisung

Wenn eine Anweisungsfolge mehrmals mit einem laufend veränderten Parameter ausgeführt werden soll (z. B. Ausdrucken von Tabellen, Berechnung von Vektoren mit laufendem Index), kann man die Laufanweisung benutzen.

Es soll z. B. eine Tabelle von Funktionswerten (Wurzel) aller geraden natürlichen Zahlen von 2 bis 100 ausgedruckt werden. Dazu gehört folgendes Programm :

```
10 FOR A=2 TO 100 STEP 2  
PRINT A,SQR(A)  
NEXT A
```

Die PRINT-Anweisung wird zuerst mit dem Wert 2 für A, dann mit A=4, A=6 usw. bis A=100 ausgeführt. In diesem Beispiel ist A die Laufvariable. Ihr Anfangswert ist 2 und der letzte verwendete Wert ist 100. Die Schrittweite (STEP) ist 2. Wenn als Schrittweite 1 gewählt wird, kann der Ausdruck 'STEP 1' entfallen.

Anfangs- und Endwert und Schrittweite können Variable oder Ausdrücke sein, vgl. die Syntaxdiagramme.

Als Beispiel soll die Zahlenfolge, die zur Bildung der Zahl e führt (vgl. Bsp. in 4.4.4), ausgedruckt werden.

```
10 LET A=1
   LET B=1
   FOR E=1 TO 2.718 STEP A
   PRINT E
   LET A=A/B
   LET B=B+1
   NEXT E
```

Die Laufanweisung wird abgebrochen, wenn die Laufvariable größer als der Endwert ist.

Die Schrittweite kann auch negativ sein. Dann wird die Laufanweisung abgebrochen, wenn die Laufvariable kleiner als der Endwert ist. Wenn die Abbruchbedingung von vornherein erfüllt ist (Anfangswert ist größer als Endwert bei positiver Schrittweite), so wird die Laufanweisung ohne Durchlauf sofort verlassen.

4.7. Unterprogrammaufruf

Oft verwendete identische Programmteile können als Unterprogramme geschrieben werden. Der Aufruf eines Unterprogrammes erfolgt mit GOSUB zellennummer. Ein Unterprogramm muß mit RETURN abschließen. Im Beispiel werden im Unterprogramm vom Bediener 3 Zahlenwerte übernommen, die den Variablen A, B, und C zugeordnet werden. Im Hauptprogramm werden mit diesen Variablen an unterschiedlicher Stelle zwei unterschiedliche Algorithmen ausgeführt.

```

10  GOSUB 30
    LET X=A*B/C
    PRINT X
    .
    .
    .
    GOSUB 30
    LET Y=A/B+C
    PRINT Y
    .
    .
    .
    END
30  PRINT 'A=';
    LET A=INP
    PRINT 'B=';
    LET B=INP
    PRINT 'C=';
    LET C=INP
    RETURN

```

4.8. Kommentaranweisung 'REM'

Nach der Anweisung REM kann in das Programm eine beliebige Textfolge aufgenommen werden, die der Kommentierung dient und nicht abgearbeitet wird. Das unterstützt eine übersichtliche Programmierung. Die nächste abzuarbeitende Anweisung muß mit einer Zeilennummer beginnen.

4.9. Programmhalt und Programmunterbrechung

Mit der Anweisung

WAIT

kann der Programmlauf unterbrochen werden, bis eine beliebige Taste betätigt wird. Danach wird das Programm normal fortgesetzt. Die Anweisung kann benutzt werden, um eine schnelle Folge von PRINT-Anweisungen sichtbar zu machen.

In dem Beispiel des Abschnittes 4.6 zum Ausdrucken der Wurzeln der Zahlen von 2 bis 100 kann besser notiert werden :

```
10 FOR A=2 TO 100 STEP 2
   PRINT A,SQR(A)
   WAIT
NEXT A
```

Erst so ist die eigentliche Tabelle sichtbar.

Die Anweisung WAIT kann auch benutzt werden, um in einem neu erstelltem Programm zwecks Test an bestimmten Stellen Haltepunkte zu setzen, um den aktuellen Bearbeitungszustand zu erfahren.

Die Anweisung

```
STOP
```

dient der Programmunterbrechung, um im Schrittest die Weiterarbeit zu ermöglichen, Variablenwerte zu kontrollieren, zu setzen usw. Näheres dazu befindet sich im Abschnitt 3.3.

4.10. Programmende

Das Ende eines Programmes sollte mit der Anweisung

```
END
```

gekennzeichnet werden. Werden im Programm Unterprogramme oder Prozeduren verwendet, so kommt es sonst zu fehlerhafter Abarbeitung, da Unterprogramme oder Prozeduren in unzulässiger Weise erreicht werden. Im einfachen Fall (die bisher verwendeten Beispiele) ist auch ein Abschluß ohne END möglich.

Die Anweisung END kann auch innerhalb eines Programmes mehrmals verwendet werden.

END bewirkt auf dem Display rechts unten die Ausschrift READY.

Nach Betätigen einer beliebigen Taste wird der Editormodus wieder angenommen und das Programm kann korrigiert oder neu gestartet werden.

5. Boolesche Ausdrücke und boolesche Wertzuweisung

Boolesche Werte werden repräsentiert von booleschen Variablen, von Vergleichsausdrücken oder von einzelnen arithmetischen Ausdrücken. Der boolesche Wert einzelner arithmetischer Ausdrücke ist 1, wenn der arithmetische Ausdruck ungleich Null ist. Ist er Null, so ist auch der boolesche Wert Null.

Boolesche Werte können über die logischen Operationen

AND	logische und - Verknüpfung
OR	logische oder - Verknüpfung
XOR	logische exklusiv-oder (antivalenz) - Verknüpfung

zu einem booleschen Ergebniswert verknüpft werden. Dabei ist

XOR gegenüber AND gegenüber OR priorisiert.

Die boolesche Funktion

NEG	logische Negation
-----	-------------------

wird von einem booleschen Wert geschrieben und negiert ihn.

NEG ist am höchsten priorisiert.

Eckige Klammern ermöglichen die vorzugsweise Abarbeitung der geklammerten Ausdrücke und damit die Änderung der Abarbeitungspriorität.

Im Beispiel

$$A \text{ OR } \text{NEG } B \text{ XOR } C$$

wird zuerst B negiert, dann mit C über XOR verknüpft und zuletzt mit A über OR verknüpft.

Im Beispiel

$$A \text{ OR } \text{NEG } [B \text{ XOR } C]$$

wird zuerst B und C über XOR verknüpft, dann wird negiert usw. Die Vergleichsausdrücke sind gegenüber den booleschen Operatoren priorisiert.

Insgesamt ergibt sich folgende Abarbeitungspriorität :

() []
* /
+ -
= >= <= > < <>
NEG
XOR
AND
OR

Boolsche Ausdrücke können in Bedingungsabfragen oder in boolschen Wertzuweisungen benutzt werden.

In boolschen Wertzuweisungen muß nach dem Schlüsselwort LET eine boolsche Variable verlangt werden :

```
          BOOLEAN B1 , B2(7)  
10      LET B1=A > B AND S=0
```

Das die boolsche Variable repräsentierende Bit wird dann gesetzt, wenn beide Bedingungen $A > B$ und $S=0$ erfüllt sind.

Sonst wird es rückgesetzt.

Alle anderen Bits des entsprechenden Bytes werden nicht verändert.

Hinweis: - Vereinbarung von boolschen Variablen im Abschnitt 9
- Boolsche Feldelemente können nicht verarbeitet werden !

6. Standardfunktionen

Folgende Standardfunktionen sind in BASIC-80 verwendbar :

SIN	Sinus
COS	Cosinus
LOG	Natürlicher Logarithmus
EXP	e - Funktion
SQR	Quadratwurzel
INT	Ganzzahlanteil
ABS	Betrag
SGN	Vorzeichen
PI	3.14159

Alle Standardfunktionen außer PI benötigen ein Argument, das nach dem Funktionsnamen in () angegeben wird.

Beispiele :

$$\text{LET } A=B*\text{SIN}(T) \quad a = b \cdot \sin t$$

$$\text{LET } A=\text{ABS}(D) \quad a = |d|$$

$$\text{LET } D=(\text{EXP}(X)-\text{EXP}(-X))/2 \quad d = \frac{e^x - e^{-x}}{2}$$

$$\text{LET } F=2*PI*R*(H+R) \quad f = 2\pi \cdot r(h+r)$$

$$\text{LET } V=4/3*PI*R*R*R \quad v = \frac{4}{3}\pi \cdot r^3$$

$$\text{LET } U1=U0*\text{EXP}(-T/TAU) \quad u_1 = u_0 e^{-\frac{t}{\tau}}$$

$$\text{LET } A=\text{SGN}(B)*\text{ABS}(C) \quad a = \text{sign } b \cdot |c|$$

$$\text{LET } A=\text{INT}(B) \quad a = \text{Intg } b$$

$$\text{LET } DB=20*\text{LOG}(U1/U2)/\text{LOG}(10) \quad db=20 \lg \frac{U_1}{U_2} = 20 \frac{\ln \frac{U_1}{U_2}}{\ln 10}$$

sign $\hat{=}$ Vorzeichen

Intg $\hat{=}$ Ganzzahlanteil

Auch in diesen Beispielen ist es sinnvoll, die errechneten Werte auf dem Display zur Anzeige zu bringen, Folgendes Programm realisiert den Ausdruck der Argumente und Funktionswerte einer Funktion in wählbaren Schritten bis zu einem gewünschten Endwert. Die Funktion sei

$$o = 4 \cdot \pi \cdot r^2$$

(Kugeloberfläche).

Mit jedem Tastendruck (bei längerem Betätigen repetierend) soll ein Wertepaar gedruckt werden. Am Anfang werden der gewünschte Endwert und die Schrittweite verlangt.

```
10 PRINT 'UUUUUU'ENDWERT=';
   LET E=INP
   PRINT
   PRINT 'SCHRITTWEITE=';
   LET S=INP
   PRINT
   FOR R=0 TO E STEP S
   PRINT 'R=';R, 'A=';4*PI*R*R
   WAIT
   NEXT R
   END
```

Verändern Sie in diesem Programm die Funktion, z. B. Berechnung einer Kreisfläche u. ä.

7. Felder: Vektoren und Matrizen

BASIC-80 ermöglicht es, Felder zu vereinbaren und zu verarbeiten. Bei der Anfangsinitialisierung und nach NEW werden keine Felder vereinbart. Diese zusätzlichen Vereinbarungen können im lokalen oder globalen Vereinbarungsteil ergänzt werden, wie es dem Abschnitt 9 zu entnehmen ist. Im Vereinbarungsteil steht nach dem Namen in Klammern die Zahl der Feldelemente. Die Felder werden nur als Vektoren mit einem Index geführt.

Feldelemente werden mit dem Feldnamen, gefolgt von einem in () stehenden Index bezeichnet. Die Zählung des Index beginnt mit 0. Der Index kann auch von dem Ganzzahlanteil eines arithmetischen Ausdruckes repräsentiert werden. Es ist zu beachten, dass der Index die im Vereinbarungsteil festgelegte obere Grenze nicht überschreitet. Ansonsten können andere Variablenwerte zerstört werden.

Beispiel : Das Feld FEL wird mit 6 Elementen vereinbart. Diesen Elementen sollen über Eingabe 6 Werte zugeordnet werden.

```
REAL FEL(6)
10 FOR A=0 TO 5
  PRINT 'FEL(';A;')=';
  LET FEL(A)=INP
  PRINT
NEXT A
FOR A=0 TO 5
  PRINT FEL(A)
NEXT A
END
```

Dieses Programm besteht aus zwei Laufanweisungen. Die erste ermöglicht die Eingabe der Werte der Feldelemente. Die zweite Laufanweisung druckt die übernommenen Werte auf dem Display zwecks nochmaliger Kontrolle aus.

Dieses Programm ist Wiederholung des Stoffes aus Abschnitt 2. Weshalb steht nach FOR A=0 TO 5 kein STEP ?

Welche Funktion haben die Semikoll in der PRINT - Anweisung ?

Bei der Vereinbarung von Feldern wird keine Zeilen- und Spaltenordnung festgelegt wie es bei Matrizen erfolgt, sondern es gibt nur eine geordnete Folge von Feldelementen, ähnlich der von Vektoren. Durch geeignete Indizierung ist es möglich, Matrizen- und Determinantenrechnungen durchzuführen. Dabei bieten sich Laufanweisungen an.

Beispiel : Spiegeln einer Matrix an der Hauptdiagonalen (Stürzen bzw. Transponieren der Matrix) . Dabei werden die Zellen zu Spalten und umgekehrt :

$$\begin{bmatrix} a_1 & b_1 & c_1 \\ a_2 & b_2 & c_2 \\ a_3 & b_3 & c_3 \end{bmatrix}^T \quad \begin{bmatrix} a_1 & a_2 & a_3 \\ b_1 & b_2 & b_3 \\ c_1 & c_2 & c_3 \end{bmatrix}$$

Im ersten Programmteil wird die Matrix folgendermaßen belegt und ausgedruckt :

$$\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

```
REAL MAT1(9),MAT2(9)
10 FOR A=0 TO 8
  LET MAT1(A)=A+1
NEXT A
CLEAR
PRINT MAT1(0);' ';MAT1(1);' '; MAT1(2)
PRINT MAT1(3);' ';MAT1(4);' '; MAT1(5)
PRINT MAT1(6);' ';MAT1(7);' '; MAT1(8)
```

In diesem einfachen Beispiel wird pro Matrixzeile eine PRINT - Anweisung verwendet.

Für größere Matrizen sind Laufanweisungen vorzuziehen.

Das Stürzen der Matrix geschieht nach folgendem Programm:

```
20 FOR I=0 TO 2
   FOR J=0 TO 2
     LET MAT2(J*3+I)=MAT1(I*3+J)
   NEXT J
 NEXT I
30 FOR A=0 TO 8 STEP 3
   PRINT MAT2(A),MAT2(A+1),MAT2(A+2)
 NEXT A
END
```

Bei der Indizierung der Matrizen ist folgende Form günstig :

MAT2(J*3+I)

-----Spaltenindex
-----Zahl der Elemente in der Zelle
 (Spaltenzahl)
-----Zeilenindex

Mathematisch wird das Matrixelement wie folgt bezeichnet :

MAT2(j,i)

In Einzelanweisungen können nur Feldelemente verarbeitet werden. Berechnungen mit kompletten Matrizen werden günstig mit zwei geschachtelten Laufanweisungen mit den Elementen ausgeführt. Wie im Beispiel verwendet bearbeitet die äußere Laufanweisung die Matrixzeilen und die innere Laufanweisung die Elemente der Zelle. Die Matrix wird hier zeilenweise abgearbeitet. Eine spaltenweise Abarbeitung wäre auch möglich.

Mit Prozeduren (vgl. Abschnitt 10) ist auch die Bearbeitung von kompletten Matrizen möglich.

8. Textverarbeitung

Mit BASIC-80 ist Textverarbeitung möglich. Es gibt Textvariable, die zusätzlich vereinbart werden müssen. Einer einfachen Textvariablen kann nur ein Einzelzeichen zugeordnet werden. Für Zeichenketten sind Textfelder erforderlich. Im Beispiel wird der Textvariablen TX die Zeichenkette 'HYPOTHENUSE=' zugeordnet. In der folgenden PRINT-Anweisung wird diese Textvariable wieder aufgerufen. Es ergibt sich das gleiche Druckbild wie im Beispiel des Abschnittes 2 :

```
STRING TX(12)
10 LET TX='HYPOTHENUSE='
PRINT TX;SQR(A*A+B*B)
```

Mit der LET - Anweisung wird einer Textvariablen eine Zeichenkette zugeordnet. Dabei muß auf der linken Seite des Gleichheitszeichens die Textvariable stehen.

Es können auch mehrere Zeichenketten verknüpft werden. Der Ausdruck nach dem Gleichheitszeichen ist dann ein Textausdruck. Er hat die gleiche Form wie bei einer PRINT - Anweisung.

Beispiel :

```
STRING W1(4),W2(8),W3(12)
10 LET W1='LAUT'
LET W2='SPRECHER'
LET W3=W1;W2
PRINT W3
PRINT W1;W2
```

Es wird zweimal untereinander das Wort LAUTSPRECHER ausgedruckt, einmal aus der Verknüpfung von W1 und W2 zu W3 und einmal aus der Verknüpfung direkt in der PRINT - Anweisung.

Zahlen oder arithmetische Ausdrücke können auch in Zeichenketten gewandelt werden:

```
STRING TX(12)
10 LET A=3.5
LET TX='A=';FORMAT 0.3;A/2
PRINT TX
```

Es wird der Text A=1.75 ausgedruckt. Der Rest der Textvariablen ist mit Leerzeichen aufgefüllt.

Auswahl von Teiltextketten

In den Beispielen wurden die Textvariablen als geschlossene Zeichenketten (Felder) behandelt. Es ist auch möglich, Einzelzeichen oder Teilzeichenketten zu verarbeiten. Mit einfacher Indizierung wird aus einer Textvariablen ein Einzelzeichen herausgelöst :

```
STRING TX(10)
10 LET TX='ABCDEFGH'
PRINT TX(3)
```

Es wird das Zeichen D ausgedruckt. Die Zählung erfolgt wie bei Vektoren mit 0 beginnend.

Teilzeichenketten werden mit zwei Indizes bezeichnet. Der erste Index bezeichnet die Stelle, an der die Teilzeichenkette beginnt. Der zweite Index gibt die Länge der Teilzeichenkette an. Im obigen Beispiel wird mit

```
PRINT TX(1,4)
```

die Zeichenkette BCDE ausgedruckt.

Es können auch Teilzeichenketten verknüpft werden :

```
STRING TR(13),TF(5)
10 LET TR='TRANSFORMATOR'
LET TF=TR(0,3);TR(5,2)
PRINT TF
```

Es wird TRAF0 ausgedruckt.

Vergleich von Zeichenketten

In einer IF - Anweisung kann der Vergleich von Zeichenketten als Bedingung benutzt werden. Auf der linken Seite des Vergleichsausdruckes kann ein beliebiger Textausdruck stehen (vgl. Syntaxdiagramme), auf der rechten Seite darf nur eine Textvariable oder eine konstante Zeichenkette stehen. Als Vergleichsoperatoren dienen die gleichen Operatoren wie für den arithmetischen Vergleich. Dabei bedeutet '<' bzw. '>' davorstehend oder nachfolgend in alphabetischer Reihenfolge.

Folgendes Beispiel druckt 8 eingegebene Worte in alphabetischer Reihenfolge aus :

```
    STRING NAME(80),TX(10),TY(10)
10 PRINT FORMAT 0.2;
   FOR A=0 TO 7
   PRINT 'NAME';A+1; '=';
   LET NAME(10*A,10)=INP
   PRINT
   NEXT A
20 LET TY=' '
   FOR B=0 TO 7
   PRINT
   LET TX='ZZZZZZZZZZ'
   FOR A=0 TO 7
   IF TY<NAME(10*A,10) AND NAME(10*A,10)<TX DO
   LET TX=NAME(10*A,10)
   DOEND
   NEXT A
   PRINT TX;
   LET TY=TX
   NEXT B
END
```

9. Vereinbarungstell, Einbindung von Assemblerfunktionen

Nach dem Kommando NEW bzw. nach Neuauwahl des BASIC - Programmes erfolgt eine Grundinitialisierung. Damit können einfache Programme (die verwendeten Beispiele) sofort notiert werden. Der Vereinbarungstell ist in den lokalen und globalen Teil getrennt. Der globale Vereinbarungstell steht zu Anfang des gesamten Programmsystems. Der lokale Vereinbarungstell gilt nur für das Hauptprogramm oder die jeweilige Prozedur (vgl. Abschnitt 10).

9.1. Lokaler Vereinbarungstell, Variablentypen

Das Hauptprogramm und jede Prozedur beginnt mit der Anweisung

```
DEF NAME
```

Dabei ist NAME eine beliebige Zeichenfolge. Bei Prozeduren wird nach dem Namen eine Liste formaler Parameter und Variabler in Klammern angegeben (vgl. Abschnitt 10).

Die auf DEF folgenden Anweisungen gehören zum lokalen Vereinbarungstell.


Der lokale Vereinbarungstell des Hauptprogramms nach der Grundinitialisierung ist im Abschnitt 3.1. bereits vorgestellt und in den Beispielen verwendet worden. Im Abschnitt 7 wurde der lokale Vereinbarungstell mit Feldvereinbarungen ergänzt.

Um zusätzliche REAL - Variable mit eigenen Bezeichnungen zu erhalten, kann der lokale Vereinbarungstell z. B. folgendermaßen erweitert werden :

```
REAL U1,U0,TAU  
REAL U2
```


Diese Variablen wurden in den Beispielen des Abschnittes 6 verwendet.

Die Anweisungen zum lokalen Vereinbarungstell haben folgende Form:



Im Vereinbarungstell dürfen keine Zeilennummern verwendet werden.

Diese Darstellung ist ein Syntaxdiagramm, wovon sich im Abschnitt 12 noch weitere befinden. Der Begriff 'name' muß dabei noch definiert werden :



Der Begriff 'buchstabe' kann ein Zeichen A...Z haben und müßte im strengen Sinne hier auch definiert werden. Jedoch ist diese Definition mit Worten umschreibbar und allgemein bekannt und soll deshalb hier entfallen. Eine 'ziffer' ist ein Zeichen 0...9.

Die Syntaxdiagramme stellen die Möglichkeiten der sinnvollen Aneinanderreihung von Zeichen dar. Dazu können alle Wege in Pfeilrichtung genommen werden. Ein 'name' muß demzufolge mit einem 'buchstabe' beginnen, gefolgt von einer beliebigen Anzahl 'buchstabe' oder 'ziffer'.

Die Syntaxdiagramme sagen nichts über die Bedeutung der Begriffe aus. Die Bedeutung der Begriffe (deren Semantik) wird hier verbal beschrieben :

'name' ist die Zeichenfolge, mit der eine Variable in einer Anweisung bezeichnet wird. Die 'ganze zahl' (nur aus einer Ziffernfolge bestehend, vgl. Abschnitt 12) gibt an, daß es sich um ein Feld handelt.

Sie bestimmt die Zahl der Feldelemente und den maximalen Wert des Feldindex.

Die Begriffe REAL, INTEGER, BOOLEAN und STRING bestimmen den Variablentyp. Die Variablentypen haben folgende Eigenschaften :

REAL

REAL - Variable sind Gleitkommazahlen. Sie haben einen Zahlenbereich von 10^{-38} bis 10^{+38} positiv oder negativ. Die Null kann dargestellt werden. Die Verarbeitungsbreite beträgt 7 Dezimalstellen.

Bsp.:

Darstellung im Format 0.4

1,234

1.234

0,125

125E-03

-3.10⁺²⁰

-300E+18

Eine REAL - Variable oder ein REAL - Feldelement belegt 4 Byte im Speicher. Intern wird eine REAL - Variable in der binären Gleitkommadarstellung mit 24 bit Mantisse und 8 bit Exponent dargestellt. Das oberste Mantissenbit ist immer 1 (die Mantisse ist normalisiert abgespeichert). An dieser Stelle wird das Vorzeichen der Gesamtzahl vermerkt. Zum Exponenten im Zweierkomplementformat wird die Konstante 80 hexadezimal addiert. Diese Zahlendarstellung ist in der Beschreibung der Gleitkommaarithmetik EPAS-80 beschrieben.

Da in der binären Zahlendarstellung manche runde Dezimalbrüche unendlich periodisch sind, können sie in der dezimalen Darstellung nicht exakt wiedergegeben werden. Insbesondere trifft das auf die Zahlen 0,1 ; 0,2 usw. zu. Es wird mit abgerundeten Werten gearbeitet und diese werden auch dargestellt. In einer Laufanweisung

```
FOR A=1 TO 100 STEP 0.1
```

wird z. B. nicht genau der Wert 100 erreicht. Ist dies erforderlich, so ist es günstiger zu programmieren :

```
FOR A=10 TO 1000 STEP 1
```

Die Laufvariable A kann intern dann mit der Bewichtung A/10 benutzt werden.

Verbleibt man im Bereich der ganzen Zahlen, so treten keine Rundungsfehler auf. Wenn die letzte Ganzzahlstelle (Wertigkeit 1) noch aufgelöst werden soll, kann man maximal die Zahl

```
8 388 607
```

verarbeiten. Das entspricht der internen Auflösung mit 24 bit.

INTEGER

INTEGER - Variablen sind Festkommazahlen. Sie umfassen einen Zahlenbereich von -32768 ... 32767 .

Intern werden sie mit 2 Byte im Zweierkomplementformat abgespeichert.

Alle Berechnungen innerhalb von BASIC-80 werden im Gleitkommaformat (REAL) ausgeführt. INTEGER - Zahlen werden deshalb vorher in REAL gewandelt. Das INTEGER - Format kann bei lokalen Variablen sinnvoll angewendet werden, wenn bei großen Feldern Speicherplatz gespart werden soll.

Ist eine INTEGER - Variable Ergebnis einer LET - Anweisung und der INTEGER - Zahlenbereich wird überschritten, so nimmt die INTEGER - Variable den größten bzw. kleinsten darstellbaren Wert an. Sie wird begrenzt. Eine Fehlerausschrift erfolgt dabei nicht.

BOOLEAN

BOOLEAN - Variable sind Einzelbits mit dem Wertevorrat

0 ; 1

Im Speicher wird nur ein Bit belegt. Auf ein Byte passen 8 BOOLEAN-Variable, die immer im Block vereinbart werden müssen. Das ist insbesondere bei der Speicherplatzaufteilung globaler BOOLEAN-Variable zu beachten. In der Reihenfolge der Variablen wird zuerst das LSB belegt. Mittels BOOLEAN - Variablen kann insbesondere eine Einzelbitverknüpfung im RAM ausgeführt werden.

STRING

STRING - Variable können Zeichen im ASCII -Code speichern. Sie belegen als Einzelvariable 1 Byte im Speicher. Wie im Abschnitt 8 dargestellt ist, sind im allgemeinen STRING - Felder notwendig, die dann entsprechend der Anzahl der Feldelemente Bytes benötigen. Neue Vereinbarungen von lokalen Variablen dürfen nur nach (darunter) den bereits getroffenen Vereinbarungen ergänzt werden. Das Streichen bereits verwendeter Vereinbarungen führt zu Anzeigefehlern.

Beim Start eines Programms mittels RUN wird am oberen Ende des von BASIC-80 benutzten RAM - Bereiches ein Speicherbereich reserviert, dessen Byte - Zahl dem Umfang des lokalen Vereinbarungsteiles entspricht. Bei jedem Prozeduraufruf (auch geschachtelt) wird ein weiterer Speicherbereich entsprechend den lokalen und formalen Variablen der Prozedur und weitere 9 Bytes darunter beansprucht. Der Aufruf einer GOSUB - Routine benötigt 9 Bytes. Dieser Bereich wird beim Verlassen der Prozedur bzw. des Unterprogrammes wieder frei. Die Abarbeitung selbst beansprucht entsprechend der Schachtelungstiefe von Klammern in arithmetischen Ausdrücken einige wenige weitere Bytes.

9.2 Globaler Vereinbarungsteil

Im globalen Vereinbarungsteil werden globale Variable, Assemblerfunktionen, Assemblerausgabefunktionen und Assembleranweisungen deklariert.

Globale Variable werden vom Programm ebenso wie lokale Variable verarbeitet. Ihre Speicheradresse wird vom Anwender festgelegt. Damit kann der Datenzugriff auch über ein Assemblerprogramm erfolgen.

Ein Block im globalen Vereinbarungsteil beginnt mit (Bsp.):

```
GLOBAL %E800
```

E800 ist eine hexadezimale Speicheradresse, auf die sich alle weiteren Eintragungen beziehen.

Die Typbezeichnungen der globalen Variablen erfolgt analog denen der lokalen Variablen mit nachgestelltem #.

```
Beispiele : REAL# A1,VE(45)
            BOOLEAN# B1,B2,B3(6)
            STRING# TEXP(70)
```

Die in diesem Block befindlichen Variablen werden ab der Speicheradresse E800 angelegt. Im einzelnen haben die Variablen folgende hexadezimalen Adressen:

A1	E800	eine REAL-Variable a 4, Byte
VE	E804	45 REAL-Feld Elemente a 4 Byte = 180 Byte = B4H
B1	E8B8, Bit 0	BOOLEAN-Variable a 1 Bit
B2	E8B8, Bit 1	
B3	E8B8, Bit 2...7	
TEXP	E8B9...E8FE	70 STRING-Variable a 1 Byte

Ein nächster Vereinbarungsblock könnte z. B. beginnen:

```
GLOBAL %E000
INTEGER# F1,F2
```

Damit werden die INTEGER-Variablen F1 und F2 auf den Adressen E000 bzw. E002H (à 2 Byte) gespeichert.

Hinweis: Bei Auswahl der Speicheradressen sind mögliche Überschneidungen mit anderen Speicherbelegungen zu beachten und zu vermeiden!

Assemblerfunktionen

Eine Assemblerfunktion ist ein Unterprogramm in Assemblersprache (U880), welche einen Wert als Ergebnis liefert. Assemblerfunktionen werden ähnlich wie Standardfunktionen verwendet. Sie können ohne Argument oder mit mehreren Argumenten auftreten. XE sei z. B. eine Assemblerfunktion ohne Argumente, die einen Analog-Digital-Wandler abfragt und seinen Wert als Ausgang liefert. In einer Anweisung wird diese Assemblerfunktion folgendermaßen verwendet:

```
LET XW=W-XE
```

Zuvor muß die Assemblerfunktion wie folgt vereinbart werden :

```
GLOBAL %EC00  
INTEGER! XE
```

Ab EC00 beginnt dieses Assemblerprogramm, das als Unterprogramm formuliert das Ergebnis im INTEGER- (Festkomma-) Format im Registerpaar HL liefert. Die genauen Softwareschnittstellen werden im nächsten Abschnitt beschrieben.

Aus programmtechnischen Gründen ist es oftmals günstig, mit einer Sprungverteilungstabelle (Folge von JMP-Befehlen) zu arbeiten. Dann können auch mehrere Assemblerfunktionen in einem Block deklariert werden.

Beispiel:

```
GLOBAL %EE00  
INTEGER! XMIN, XMAX, SOLL  
BOOLEAN! STEU1, STEU2  
INTEGER! VEN
```

Ab der Adresse EE00 stehen dann die Sprünge zu den eigentlichen Programmen :

```
JMP MIN  
JMP MAX  
JMP SOL  
JMP ST1  
JMP ST2  
JMP VEN
```

Assemblerausgabefunktion

Eine Assemblerausgabefunktion ist ein Unterprogramm in Assemblersprache (U880), die einen Wert als Eingang benötigt. VEN sei z. B. eine Assemblerausgabefunktion, die ein Stellglied (z. B. Ventil) ansteuert. Die Assemblerfunktion wird in BASIC folgendermaßen verwendet :

```
LET VEN=I+K*(W-XE)
```

Das Ergebnis des arithmetischen Ausdruckes wird der Ausgabefunktion in den CPU-Registern übergeben. Vereinbart werden Assemblerausgabefunktionen genauso wie Assemblerfunktionen mit nachgestelltem '?' (anstatt !).

```
GLOBAL %EE0F  
INTEGER? VEN
```

Assemblerausgabefunktionen können sich nur auf der linken Seite einer LET-Wertzweisung befinden.

Assembleranweisungen

Eine Assembleranweisung ist ein Unterprogramm in Assemblersprache ohne zusätzliche Schnittstellen. In BASIC wird eine Assembleranweisung wie folgt verwendet (Bsp.) :

```
REGLER
```

Vereinbart wird diese Assembleranweisung :

```
GLOBAL %EE12  
DCL REGLER
```

Wenn die Anweisung REGLER erreicht wird, wird ein Assemblerprogramm auf der Adresse EE12 (hexadezimal) gestartet.

Die Einbindung von Assemblerprogrammen in BASIC-80 ermöglichen die Ausnutzung des gesamten Befehlssatzes des Prozessors U880 und damit eine spezifische hardwarenahe Programmierung.

9.3. Softwareschnittstellen der Assemblerprogramme

Assembleranweisungen:

INPUT: DE = Zeiger auf die nächste Position des Zwischencodes BASIC, die abgearbeitet werden soll

 IY = arithmetischer Stackpointer, arbeitet decrementierend

 IX = Zeiger auf lokale Variable (im Stack)

OUTPUT: DE = Zeiger auf nächste abzuarbeitende Position im Zwischencode, ggf. bei Parameterübernahme verändert

 IY = unverändert!

 IX = unverändert!

 Cy = 0 : keine Fehlermeldung

 Cy = 1 : Syntaxfehler führt zur Unterbrechung des BASIC-Programmes

Parameter können den Assembleranweisungen nach dem Namen in Klammern () angegeben werden, z. B. :

 REGLER(7.5,W-XE)

Nach Aufruf des Assemblerprogrammes zeigt Register DE im Zwischencode die '('. Da IX und IY auf die lokalen Variablen bzw. den Stack zeigen, kann mit folgendem Assemblerprogramm eine Parameterübernahme erfolgen:

PUE:	INC DE	;Überlesen (
	CALL TRM	;Ermitteln des ersten Parameters
	RC	;Rücksprung falls Syntaxfehler
	INC DE	;Überlesen des Kommas
		;das Ergebnis des ersten arithmetischen
		Ausdruckes steht in den Registern
		BCHL im Gleitkommaformat
	CALL FKI	;Invertierung in Festkomma, Register
		HL, mit Überlaufbegrenzung
	PUSH HL	;merken
	CALL TRM	;zweiter arithmetischer Ausdruck nach
		BCHL
	JRC FEH	;Rücksprung falls Syntaxfehler
	CALL FKI	;Festkomma in HL
	POP BC	;gemerkter Wert
	INC DE	Überlesen)
	RET	;Rücksprung zum aufrufenden Programm
FEH:	POP HL	;Stack korrigieren
	RET	;Rücksprung bei Fehler mit Cy=1
TRM:	EXT 5D7AH	;Programme sind Bestandteil des BASIC-
FKI:	EXT 5B94H	;interpreters

Die Einbettung in das Gesamtprogramm erfolgt nach folgendem Schema :

REG: CALL PUE	; Parameterübernahme
RC	; Rücksprung bei Fehler
PUSH DE	; Retten der Register, DE wurde bei der Parameterübernahme incrementiert und zeigt auf die nächste Anweisung
PUSH IY	
PUSH IX	
	; Abarbeiten des gewünschten Algorithmus
	- im Doppelregister BC steht im Zweierkomplement der 1. Parameter
	- im Doppelregister HL der 2. Parameter
POP IX	; Wiederherstellung der ursprünglichen
POP IY	; Belegung der Register
POP DE	
XOR A	; Cy=0, zur Kennzeichnung der fehlerfreien Abarbeitung
RET	; Rücksprung zum BASIC-Interpreter

Assemblerausgabefunktionen

Es gelten die gleichen Schnittstellen wie für Assembleranweisungen. Zusätzlich gelten folgende INPUT-Parameter :

INPUT für

REAL?: BCHL =Gleitkommazahl als Ergebnis des arithmetischen Ausdruckes der LET-Anweisung

INTEGER?: HL=Festkommazahl als Ergebnis des arithmetischen Ausdruckes der LET-Anweisung

BOOLEAN?: A=00 oder FF als Ergebnis des booleschen Ausdruckes der LET-Anweisung

STRING?: In dem STRING-Unterprogramm muß mit den INPUT-Parametern HL=Textpufferadresse und C=Textpufferlänge das im BASIC-Interpreter enthaltene Programm TTX aufgerufen werden!

Assemblerfunktionen

Es gelten die gleichen Schnittstellen wie für Assembleranweisungen.
Zusätzlich müssen folgende OUTPUT-Parameter realisiert werden :

OUTPUT für

REAL!	BC HL=Gleitkommazahl
INTEGER!	HL=Festkommazahl
BOOLEAN!	A=00 oder FF

Für die STRING-Funktion werden INPUT-Parameter übergeben :

INPUT für STRING! :

HL=Adresse des Textpuffers
(IY)=Verfügbare Länge des Textpuffers
(IY+1)=Formatbyte

OUTPUT für STRING! :

Im Assemblerprogramm kann der Textpuffer nachfolgend mit beliebigen Zeichen in der ASCII-Codierung (vgl. Handbuch MC 80) gefüllt werden. Dabei muß (IY) beim Einschreiben jedes Zeichens decrementiert werden. (IY) darf den Wert 1 nicht unterschreiten! HL muß bei Verlassen des Programmes das nachfolgende Byte zeigen!

Beispiel einer STRING-Funktion :

Die STRING-Funktion soll alle Zeichen mit den Codierungen 20H...5FH erzeugen. Damit kann gleichzeitig der ASCII-Code überprüft werden. In Abhängigkeit von der Textpufferlänge der Operation in BASIC wird nur eine begrenzte Zeichenzahl erzeugt. (Programm ab EC00 ablegen).

```
ASC:  LD A,20H      ; erster Zeichencode
AS1:  DEC (IY)     ; decrement. und Testen der Textpufferlänge
      JRNZ AS2     ; nicht Null
      INC (IY)     ; Null: Incrementieren auf 1
      XOR A       ; Beenden des Programmes vorzeitig ohne
      RET        ; Fehlermeldung
AS2:  LD (HL),A    ; ASCII-Zeichen in Textpuffer eingeben
      INC HL      ; nächste Textpufferposition
      INC A       ; nächstes ASCII-Zeichen
      CMP 60H     ; Abbruchkriterium
      JRNZ AS1    ; weiter, wenn 60H noch nicht erreicht wurde
      XOR A       ; Rücksprung ohne Fehlermeldung
      RET
```


In BASIC wird dieses Programm folgendermaßen verwendet :

```
GLOBAL %EC00 (Startadresse)
STRING! ASCII
GLOBAL %E000
STRING# TEXT(64)
DEF HAUPTPROGRAMM
...usw.
10 LET TEXT=ASCII
REM AB DER ARDRESSE E000 STEHEN JETZT DIE BYTES 20...5F
20 PRINT TEXT(0,16)
PRINT TEXT(16,16)
PRINT TEXT(32,16)
PRINT TEXT(48,16)
REM NUN WERDEN ALLE ASCII - ZEICHEN AUSGEDERUCKT
30 END
```

Vereinbaren Sie für die STRING-Variable eine kürzere STRING-Länge
STRING TEXT(30)

Löschen Sie im RAM-Modus (MC 80 - Bedienhandbuch) den Bereich
ab E000 und wiederholen Sie die Programmausführung.

Wegen der Korrektur im Vereinbarungstell muß die Zeile 10 nochmals
übersetzt werden : Anwahl der Zeile, Kursorbewegung nach links
oder rechts,
'ENTER-Betätigung!

Arbeitet die STRING-Funktion richtig?

Die folgende Tabelle gibt über verwendbare Unterprogramme im BASIC-Interpreter Auskunft :

Name	Adr. (hex.)	INPUT	OUTPUT	Bedeutung
TRM	5D7A	DE	DE BCHL Cy	Abarbeitung eines arithm. Ausdruckes Zeiger auf Zwischencode, wird increment. Gleitkommazahl als Ergebnis gesetzt bei Syntaxfehler
TTX			DE (HL) Cy	Abarbeitung eines STRING-Ausdruckes Zeiger auf Zwischencode Adresse des Textempfangspuffers Länge des Textempfangspuffers aus Zwischencode ermittelter Text gesetzt bei Syntaxfehler
VAD	5C71	DE	DE HL B C Cy	Ermitteln der Adresse einer Variablen Zeiger auf Zwischencode Variablenadresse Variablentyp Bits b4b3: 00 REAL 01 INTEGER 10 BOOLEAN 11 STRING STRING-Länge oder BOOLEAN-Maske gesetzt bei Syntaxfehler
FKI	5B94	BCHL	HL	Festkommaintervallierung Gleitkommazahl Festkommazahl, begrenzt
ZSS	565D	DE (IY)	DE (IY) (DE)	Zeichen in Textpuffer einschreiben Textpufferzeiger noch verfügbare Textpufferlänge, (IY)≠0! Text
ZGL	5803	DE (DE)	DE C HL	ganze Zahl vom Textpuffer lesen Textpufferzeiger enthält Ziffernfolge Zahl, positiv
ZKL	5831	DE (DE)	DE CHL	gebrochene Zahl lesen (Dezimalbruch) Textpufferzeiger enthält Dezimalpunkt und Ziffernfolge Zahl, MSB Wertigkeit 2^{-1}
GZL	586C	DE (DE)	DE BCHL	Gleitkommazahl lesen Textpufferzeiger enthält allg. Zahl mit VZ, z.B. +12.34E-5 Zahl im Gleitkommaformat

Name	Adr. (hex.)	INPUT	OUTPUT	Bedeutung
ZGS	58E1	DE A HL	DE A (DE)	ganze Zahl auf Textpuffer schreiben Textpufferzeiger Stellenzahl 0...5, 0 : nur notwendige Stellen ganze positive Zahl Ziffernfolge
GZS	567B	DE (1Y) BC HL	DE (DE)	Gleitkommazahl ausschreiben Textpufferzeiger Formatbyte vgggebbb---Nachkommastellenzahl <div style="margin-left: 100px;"> Tab.-gerecht Stellenzahl ganzer Teil, 000:Ing.Format kein Exponent Vgl.Formatanw. </div> Zahl im Gleitkommaformat Zahl in Zeichenfolge

Tabelle 9.1 : Unterprogramme im BASIC-Interpreter

9.4. Aufruf von BASIC-Programmen aus Assemblerprogrammen
 BASIC-Programme können als eigenständige Tasks außerhalb des BASIC-Editors verwendet werden.

Wie im Abschnitt 3.1 erläutert kann mit dem Kommando PUT ein geschriebenes BASIC-Programm auf einen Pufferspeicher ausgelagert werden und dort als Assembleranweisung in BASIC aufgerufen werden. Ein getestetes lauffähiges Programm kann auch aus der Assemblerebene (Maschinenebene des Prozessors) aufgerufen werden. Dem mit PUT ausgelagertem Programm muß dann übergeben werden :

INPUT IY=arithmetischer Stackpointer
 Ein genügend großer RAM-Bereich mit IY als obere Grenze muß belegbar sein. Größe des RAM-Bereiches richtet sich nach der Zahl der lokalen Variablen (vgl.9.1), nach den aufgerufenen Prozeduren (vgl.10) und der Schachteltiefe von Klammern,
 DE=Zeiger auf BASIC-Zwischencode mit aktuellen Variablen und Parametern, wenn eine Liste formaler Variabler und Parameter Bestandteil des abzuarbeitenden BASIC-Programmes ist, vgl. Abschnitt 10.

Beispiel : Ein Task soll den Wert wählbarer Variabler dimensioniert mit Maßeinheit auf dem Display sichtbar machen. Der Task wird über einen Interrupt aller 100 Millisekunden zyklisch aufgerufen.

Die Variablen stehen ab den Speicheradressen E600.

Das zugehörige BASIC-Programm lautet folgendermaßen :

```

GLOBAL %E600
INTEGER' SOLL,IST,P,XMAX,X,Z
DEF
REAL A,B,C
PRINT FORMAT !3.1!;
10 DPL 0,0 'SOLLWERT=';SOLL/256+100;' GRAD'
DPL 0,25 'P=';P/64;' KW'
IF P/64 >100 DO
DPL 0,31 '*'
ELSEDO
DPL 0,31 ' '
DOEND
DPL 1,0 'ISTWERT=';IST/256+100;'GRAD'
END
  
```

Beispielwerte : Wählen Sie den RAM-Modus aus der Menütabelle aus und geben Sie ab der Adresse E600 die folgenden Werte ein : 00 34 00 20 00 10

Aus Gründen der Einsparung von Speicherplatz entfällt der Kommentierung dienende Name. Der Begriff

DEF

muß stehenbleiben!

Das Programm realisiert die Anzeige eines Soll- und Istwertes sowie einen Leistungsverbrauch. Übersteigt die Leistung den Wert 100 kW, so wird ein * ausgeschrieben. Die INTEGER-Variablen SOLL und IST haben 8 Bits nach dem Komma, der Nullpunkt liegt bei 100 Grad.

Die Variable P hat 6 Bit nach dem Komma. Das ist aus dem Ausdruck zur Dimensionierung der Anzeigegröße erkennbar.

Das Programm kann getestet werden, indem die Variablen im Direktmodus mit Werten belegt werden. Dazu wird nach DEF die Anweisung STOP eingefügt, vgl. Abschnitt 3.3.

Nach erfolgtem Test soll das Programm auf die Adresse E000 ausgelagert werden. Dazu wird in den globalen Vereinbarungsteil hinzugefügt :

```
GLOBAL %E000  
DCL ANZEIGE
```

Mit dem Kommando

```
PUTD ANZEIGE
```

wird das Programm dann auf E000 ausgelagert. Voraussetzung ist, daß der Speicher ab E000 frei (mit FFH belegt) ist. Dies ist nach INIT der Fall.

Das folgende Assemblerprogramm führt zum zyklischen Aufruf dieses Programmes aller 20 Millisekunden.

```

ANI : LD HL;INT           ; Initialisierungsprogramm
      LD (0D08H), HL      ; Interruptstartadresse
      LD A, 08H
      OUT 0F8H           ; Interruptvektor an CTC auf MBI
      LD A, 0A7H
      OUT 0F8H           ; Betriebsart Zeitgeber
      LD A, 192
      OUT 0F8H           ; Zeitkonstante
      RET

INT:  EI                 ; Interruptprogramm
      PUSH AF            ; alle Register retten
      PUSH HL
      PUSH DE
      PUSH BC
      PUSH IX
      PUSH IY
      LD HL, 0E000H      ; Startadresse des BASIC-Programmes
      LD A, (HL)
      CMP 0FFH          ; steht dort das Programm bereit?
      LD IY, 0E600H     ; arithmetischer Stack bis E5FFH
      CANZ 0E000H       ; Aufruf des BASIC-Programmes, wenn
                        ; es bereit steht
      JPC FEH           ; ggf. Fehlerbehandlung
FEH:  POP IY             ; alle Register zurückholen
      POP IX
      POP BC
      POP DE
      POP HL
      POP AF
      RETI

```

Das Initialisierungsprogramm muß einmalig aus dem Betriebssystem aufgerufen werden. Danach läuft in der Interruptebene zyklisch der Anzeigeinterrupt ab. Über das Menü ist das RAM-Programm anwählbar. Schreiben Sie damit auf die Adressen ab E600 Bytes ein und beobachten Sie die Anzeige!

Auch BASIC ist anwählbar. Wählen Sie BASIC an und holen Sie mittels

GET ANZEIGE

das Programm in das BASIC-System zurück. Der Interrupt müßte an sich zuvor gesperrt werden! Da der Quellbereich E600 aber bei GET gelöscht wird und dieses im Interrupt ausgetestet wird, kann hier darauf verzichtet werden.

Überlegen Sie, wie man den Interrupt sperren müßte!

Das Programm läßt sich jetzt editieren (verändern). Nachdem Sie es wieder getestet haben, lagern Sie es jetzt mit

PUTD ANZEIGE

wieder aus.

Der Interrupt läuft mit dem geänderten Programm wieder an.

10. Prozeduren und Funktionen

Im Abschnitt 4.7 wurde bereits auf die Verwendung von Unterprogrammen hingewiesen, wenn eine gleiche Befehlsfolge mehrmals im Programm auftritt. Eine weitere Möglichkeit stellt dabei die Anwendung einer Prozedur dar. Gegenüber Unterprogrammen haben die Prozeduren Vorteile bei der Parameterübergabe.

In einem Programm tritt z. B. die Multiplikation von Matrizen mehrmals auf. Ein Unterprogramm (GOSUB) kann man nur für Matrizen, die bei der Programmierung festgelegt werden, schreiben.

Sollen verschiedene Matrizen multipliziert werden, so sind die Werte dieser Matrizen zuvor auf jene Matrizen, mit denen im Unterprogramm gearbeitet wird, zu laden. Die Werte der Ergebnismatrix des Unterprogrammes müssen wiederum auf die gewünschte Ergebnismatrix umgeladen werden. Wenn ein gleicher Algorithmus mit verschiedenen Variablen und Parametern ausgeführt werden soll, ist bei Verwendung eines Unterprogrammes also jedesmal ein Umladen der Werte notwendig.

Prozeduren arbeiten nach außen hin mit formalen Variablen und Parametern. Diesen formalen Variablen und Parametern werden beim Aufruf die aktuellen Adressen und Werte zugewiesen.

Im Beispiel entfällt das Umladen der Matrixwerte, der Prozedur werden lediglich die drei Adressen der Argumenten- und Ergebnismatrizen als aktuelle Variable übergeben.

Beispiel : Im Hauptprogramm werden die beiden Matrizen MA und MB mit konstanten Werten geladen. Danach folgt der Prozeduraufruf (CALL). Die Ergebnismatrix MC wird anschließend auf dem Display dargestellt.

Die Ausgangsmatrizen werden dabei geladen mit :

$$MA = \begin{bmatrix} 1 & 2 & 3 & 4 \\ 5 & 6 & 7 & 8 \\ 9 & 10 & 11 & 12 \end{bmatrix} \quad \text{und} \quad MB = \begin{bmatrix} 10 & 11 \\ 12 & 13 \\ 14 & 15 \\ 16 & 17 \end{bmatrix}$$

Die Ergebnismatrix hat die Dimension $MC(3,2)$ und entsteht durch Multiplikation jeweils aller Zeilen und Spalten der Ausgangsmatrizen ($1 \cdot 10 + 2 \cdot 12 + 3 \cdot 14 + 4 \cdot 16 = MC(0)$ usw.).


```

REAL MA(12),MB(8),MC(6)
10 FOR A=0 TO 11
LET MA(A)=A+1
NEXT A
20 FOR A=0 TO 7
LET MB(A)=A+10
NEXT A
30 CALL MAMU(MC,MA,MB,3,4,2,)
40 FOR A=0 TO 5 STEP 2
PRINT MC(A),MC(A+1)
NEXT A
END

```

Der Prozedur mit dem Namen MAMU werden nachstehend in () die drei Variablennamen der Matrizen MC, MB und MA sowie die Ausdehnung der Ausgangsmatrizen 3,4,2 angegeben. Die Matrix MA hat die Ausdehnung (3,4) und MB (4,2). Da die Spaltenzahl der ersten Matrix mit der Zellenzahl der zweiten Matrix übereinstimmen muß, ist im Beispiel die 4 nur einmal angegeben.

Die Prozedur wird unmittelbar nach dem END des Hauptprogrammes notiert :

```

DEF MAMU(#C,#A,#B,R,S,T)
REAL I,J,K,M
100 FOR I=0 TO R-1
110 FOR K=0 TO T-1
LET M=0
120 FOR J=0 TO S-1
LET M=M+A(I*S+J)*B(J*T+K)
NEXT J
LET C(I*T+K)=M
130 NEXT K
NEXT I
140 RETURN
SUBEND

```

Im Prozedurkopf steht nach DEF der Prozedurname. Danach folgt in () die Liste der formalen Variablen und Parameter. Dem Prozedurkopf darf keine Zeilennummer vorangestellt werden!

Die drei formalen REAL-Variablen, die in der Prozedur als Matrizen verwendet werden, sind mit vorangestelltem # gekennzeichnet.

Die Argumentenmatrizen werden dabei geladen mit

$$MA = \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

Zur Kennzeichnung formaler Variable wird verwendet :

- # für formale REAL-Variable
- % für normale INTEGER-Variable
- & für formale BOOLEAN-Variable
- ⌘ für formale STRING-Variable (Dollarzeichen)

Diese Sonderzeichen sind dem Namen vorangestellt und gehören selbst nicht mit zur Variablenbezeichnung. Die formalen Variablen werden in der formalen Liste des Prozedurkopfes nicht als Vektoren geführt.

Nach den formalen Variablen mit dem vorgestellten Vorsatzzeichen können in der formalen Liste weitere Variablenbezeichnungen folgen. Diese Variable sind immer REAL-Variable. Ihnen wird beim Prozeduraufruf in der aktuellen Liste ein Wert zugeordnet, der von einem beliebigen arithmetischen Ausdruck repräsentiert werden kann. Diese Variable werden formale Parameter genannt.

Die formalen Parameter existieren nur in der Prozedur. Ihnen wird beim Aufruf ein aktueller Wert zugewiesen. Diese Aufrufart wird 'call by value' genannt. Im Gegensatz dazu werden den formalen Variablen aktuelle Variable zugewiesen, die gelesen und beschrieben werden können. Diese Aufrufart heißt 'call by reference'.

Über die formalen Variablen kann die Prozedur dem aufrufenden Programm Ergebnisse übermitteln. Das ist sonst nicht möglich.

Die formale Liste des Prozedurkopfes muß mit der aktuellen Liste des Prozeduraufrufes in Anordnung und Art der Variablen und Parameter genau übereinstimmen. Ansonsten liegt ein Syntaxfehler bei der Abarbeitung vor.

Dem Prozedurkopf muß ein lokaler Vereinbarungsteil folgen. In ihm werden lokale Variable definiert, die nur in dieser Prozedur gültig sind. Außer auf die in der Prozedur selbst definierten Variable kann in der Prozedur noch auf alle globalen Variable zurückgegriffen werden, die am Anfang des Programmsystems definiert wurden (vgl. Abschnitt 9). Auf die lokalen Variablen des aufrufenden Programms kann nur indirekt über die formalen Variablen zugegriffen werden. Eine direkte Zugriffsmöglichkeit besteht nicht.

Wenn die formale Liste im Prozedurkopf entfällt, so muß der lokale Vereinbarungsteil mit REAL beginnen. Prozeduren sind beliebig schachtelbar. In jeder Prozedur kann wiederum ein Prozeduraufruf, auch der eigenen Prozedur erfolgen.

Da bei jedem Prozeduraufruf ein weiterer lokaler Variablenbereich im arithmetischen Stack belegt wird, sind Prozeduren prinzipiell wiedereintrittsfähig (mehrfach geschachtelt aufrufbar, ohne Daten zu zerstören). Bei Verlassen der Prozedur wird der belegte Stackbereich wieder freigemacht. Damit verschwinden alle lokalen Variablenzuweisungen.

Funktionen

Als Prozedur geschriebene Funktionen sind ein Spezialfall einer Prozedur. Die Funktion übermittelt direkt einen Ergebniswert. Dazu muß am Ende der Funktion vor dem RETURN notiert werden

FN=arithmetischer Ausdruck

Beispiel: Die Formel zur Ermittlung des Widerstandes einer Parallelschaltung von Widerständen soll als Funktion mit 2 Parametern realisiert werden .

Die Bezeichnung der Funktion muß mit FN ... beginnen. Einzeilige Formulierung von Funktionen sind nicht möglich. Die Verallgemeinerung des Aufbaues der Prozeduren und Funktionen sind den Syntaxdiagrammen zu entnehmen.

```
REAL R1,R2,RP
10 PRINT 'R1=';
LET R1=INP
PRINT 'R2=';
LET R2=INP
PRINT
LET RP=FNPAR(R1,R2)
PRINT 'RP=';RP
GOTO 10
END
DEF FNPAR(A,B)
100 FN=A*B/(A+B)
RETURN
FNEND
```

11. Weitere Beispiele

Verarbeitung eines Datensatzes

Ein Meßplatz diene zur Aufnahme von Filterresonanzkurven. Dabei ermöglicht eine spezielle Anordnung (Frequenzgenerator, Digitalvoltmeter oder A/D-Wandler) die Eingabe von frequenzabhängigen Spannungswerten in den RAM des MC 80, die den Resonanzwiderstandsverlauf über die Frequenz beschreiben. Diese Werte liegen in einem Festkommaformat mit 8 Bit, nur positiv vor. Das bedeutet, das Byte 00 repräsentiert den Wert 0 und FFH repräsentiert den größeren Wert. Dieser Datensatz soll so bearbeitet werden, das auf dem Display die Durchlaßkurve im logarithmischen Maßstab erscheint (dB) und u.a. die Resonanzfrequenz ermittelt wird.

Um den Datensatz in BASIC zu lesen, sind Assemblerprogramme erforderlich. Diese werden mit dem Editor (MC 80) ab der Adresse EE00H notiert :

```
JMP:  JMP RST          ; Sprungvertellertabelle
      JMP REA
      JMP WRI

RST:  LD HL, 0E00H     ; RESTORE - Funktion:
      LD (0EDFEH),HL   ; Zeiger auf Datensatzanfang stellen
      XOR A           ; Cy=0 als Rücksprungbedingung
      RET

      ; Lesen eines Wertes fortlaufend (READ)
REA:  LD HL, (0EDFEH)  ; Zeiger auf Datensatz
      LD A, (HL)       ; Wert lesen (ein Byte)
      INC HL          ; Zeiger auf den nächsten Wert setzen
      LD (0EDFEH),HL  ; Zeiger zurückladen
      LD L, A
      LD H, 0          ; Übergabe des Wertes als INTEGER-Zahl in HL
      XOR A           ;
      RET
```

```

; Schreiben auf den Datensatz (zur Datenerzeugung)
WRI: LD A,L ; Wert ist niederwertiger Teil der INTEGER-Zahl in HL

LD HL,(0EDFEH) ; Zeiger auf Datensatz
LD (HL),A ; Wert ablegen
INC HL ; Zeiger rücken
LD (0EDFEH),HL
XOR A
RET

```

Da das Beispiel vorerst simuliert wird, wird der Datensatz in einem BASIC-Programm selbst erzeugt. Dazu ist auch das Assemblerprogramm WRI notwendig.

Zuerst wird der globale Vereinbarungsteil ergänzt :

```

GLOBAL %EE00
DCL RESTORE
INTEGER! READ
INTEGER? WRITE
GLOBAL %F400
INTEGER# DATEN
DEF HAUPTPROGRAMM
...usw.

```

Die drei Assemblerprogramme befinden sich mit einer Sprungvertindertabelle ab der Adresse EE00H.

Ab der Adresse F400H wird ein Vektor aufgebaut.

Für den Leitwert eines einfachen Parallelschwingkreises mit R, L und C gilt :

$$Y(j\omega) = \frac{1}{R} + j\omega C + \frac{1}{j\omega L}$$

w ist die Kreisfrequenz $w=2\pi f$.

Als Beispiel wird eingesetzt:

$$C = 500\text{pF}$$

$$L = 180\mu\text{H}$$

$$R = 100\text{k}\Omega$$

Der Betrag des Leitwertes errechnet sich zu

$$|Y(\omega)| = \sqrt{\frac{1}{R^2} + \left(\omega C + \frac{1}{\omega L}\right)^2} = \sqrt{\text{REAL}(Y)^2 + \text{Imaginär}(Y)^2}$$

Mit folgendem Programm erreicht man die Erstellung des Datensatzes mit Frequenzen von 450 kHz bis 600 kHz in Schritten von 0.1 kHz:

```

        REAL FRES ,RRES
110    RESTORE
        GRAPH
        LET R=100E+3
        LET C=500/1E+12
        LET L=180E-3
        LET FRES=1/2/PI/SQR(L*C)
        PRINT FRES
120    FOR F=450E+3 TO 600E+3 STEP 100
        LET W=2*PI*F
        LET B=1/SQR(1/R/R+(W*C-1/W/L)*(W*C-1/W/L))
        LET GRA=100*B/R
        LET WRITE=B/500
        DPL 7,0 B
        NEXT F
130    END

```

Nebst der Abspeicherung im Datensatz ab der Adresse E000H werden die errechneten Werte noch graphisch und numerisch angezeigt.

Das nächste Programm zur Bearbeitung des Datensatzes kann davor notiert werden und wird demzufolge zuerst abgearbeitet :

```

10    RESTORE
        LET RRES=0
20    FOR F=0 TO 1500
        LET B=READ
        IF B > RRES DO RRES=B
        LET FRES=F
        DOEND
        NEXT F
30    PRINT RRES ,FRES/10+450E+3
40    RESTORE
        GRAPH
        LET J=0
        LET K=0
        LET C=20/LOG(10)

```

```

50  FOR F=0 TO 1500
    LET B=READ
    LET DATEN(K)=10*C*LOG(B/RRES)
51  IF J < 5 DO J=J+1
    ELSE DO J=0
    DPL 7,0 DATEN(K)/10,B/RRES
    LET GRA=DATEN(K)/2+100
    DOEND
52  LET K=K+1
    NEXT F
    END

```

Im ersten Teil dieses Programmes wird durch Suche des Maximums aus dem Datensatz die Resonanzfrequenz und der zugehörige Resonanzwiderstand festgestellt.

Der zweite Teil ab der Zeile 40 dient dann der logarithmischen Darstellung der Werte des Datensatzes, normiert auf RRES. Die graphische Darstellung und die Anzeige werden nur bei jedem fünften Wert bedient.

Gleichzeitig werden alle Werte normiert logarithmisch in einem Vektor DATEN zwecks Weiterverarbeitung gespeichert. Auf den Vektor ist ein willkürlicher Zugriff möglich, im Gegensatz zum Datensatz, auf den nur über den laufenden Zeiger zugegriffen werden kann.

Die graphische Anzeige (Variable GRA) muß dimensioniert werden, da nur Werte zwischen -128 und 127 sinnvoll sind.


Der INTEGER-Vektor speichert die Werte in einer Auflösung von 0.1 dB ab.

12. Syntaxdiagramme


In den Syntaxdiagrammen wird die Zulässigkeit der Kombination von Bezeichnungen definiert. Sie erklären nicht die Wirkung der Bezeichnungskombinationen im Programm (Semantik und Pragmatik).

Feste Bezeichnungen, die genauso geschrieben werden müssen, werden mit Großbuchstaben bzw. den entsprechenden Sonderzeichen dargestellt. Bezeichnungen in Kleinbuchstaben stehen für einen Begriff, der in einem weiteren Syntaxdiagramm definiert wird. Dabei kann dieser Begriff selbst wieder in der Definition enthalten sein (rekursive Definition).

Unterstrichene Bezeichnungen in Kleinbuchstaben sind Namen, die hier vereinbart werden. Diese Namen beginnen mit einem Buchstaben A...Z und enthalten danach eine beliebige Folge von Ziffern und Buchstaben :









Die Bezeichnungen können in Pfeilrichtung kombiniert werden. Die Syntaxdiagramme sind von oben nach unten, von der Gesamtstruktur zum Detail gegliedert (top down).




Im Anweisungsblock sind alle vereinbarungen des zugehörigen lokalen vereinbarungsteiles, des globalen vereinbarungsteiles sowie der prozedurname und fname gültig.

globaler vereinbarungsteil :






arithmetische konstante




lokaler vereinbarungstell:


1) 2)




formale parameterliste:




- anweisung :




- anweisungen :



- assembleranweisungen , die initialisiert werden :



- korrespondierende aktuelle parameterliste :



zahl und reihenfolge muß der formalen parameterliste der aufgerufenen prozedur bzw. funktion entsprechen!


- korrespondierende aktuelle übergabeliste :

richtet sich nach den gegebenheiten des assemblerprogramms, erfolgt dort keine übernahme, so ist die liste leer.


- Integerausgabefunktion, die initialisiert wird :




- realfunktionen, die initialisiert werden :




stringfunktion, die initialisiert wird




arithmetischer ausdrück



boolescher ausdrück :



textausdruck :



V-5-2 Mn 44286 1046 0

Erstellt am: 16.04.2019

von Elmar Klinder